
SPL: an overview
This chapter sketches the basic features of SPL and compares them to their C 
equivalents.

A sample program
Before discussing the features of SPL, we present a little program that gives the 
flavour of the language. As is the usual practice, we shall start the SPL tour with the 
infamous “Hello, world” program.

[1] #include "io.sph"
[2] using spl;
[3]
[4] procedure main() public
[5] {
[6]     Stdout::print("Hello, world!!!\n");
[7] }

(the line numbers are not really part of the SPL program, and are given here just for 
convenience reasons. In many of the subsequent examples, line numbers shall be 
omitted).

The following sections describe the above program in some details.

#include “io.sph”
The first line of the program is not really an SPL construct – it’s an SPL preprocessor 
directive. It tells the compiler that the contents of the standard header file io.sph
shall be processed at this point. The exact location of this file is platform-specific (so 
the SPL compiler will search for this file differently on different platforms), but the 
actual contents of the io.sph file is the same – it declares a number of routines used 
for I/O, specifically, the procedure spl::Stdout::print.

using spl;
This using directive tells the compiler that all identifiers prefixed with “spl::” 
can be used without having to write the prefix each time they are used. The prefix 
“spl::” is common to all SPL standard library elements (such as the 
spl::Stdout::print procedure), so the using declaration effectively brings 
all SPL standard library elements defined in the standard header file io.sph into the 
current scope, allowing to use SPL standard library elements without writing the 
“spl::” prefix for each use. Without this using directive, the line [6] would have 
been written as:

spl::Stdout::print("Hello, world!!!\n");

Procedure main
An SPL program consists of procedures. One of these – called main – is quite special 
– an execution of an SPL program normally starts from this procedure (there are 
exceptions to this rule, see the chapter “Hosted and freestanding environments”). 

specified size and are platform-independent. SPL also provides a platform-dependent 
integer type, which represents signed integer values of the platform-specific size. 
These five types are complemented by five unsigned integer types – cardinal*1, 
cardinal*2, cardinal*4, cardinal*8 and cardinal. This strategy differs 
radically from C, where nothing is known about the precision of integer types.

Similarly to integer types, SPL provides two real types – real*4 and real*8, both 
of which are platform-independent. SPL also provides a platform-dependent real
type, which represents real values of the platform-specific size.

Unlike C, where character type is emulated using small integers, SPL provides three 
dedicated character types – character*1, character*2, and character*4. 
These types represent, correspondingly, ANSI, Unicode BMP and ISO-10646 
characters. SPL also provides a platform-dependent character type, which 
represents characters of the platform-specific size.

The boolean type is another feature of SPL absent in C. It represents Boolean truth 
values.

An enumerated type is a type whose values are specified (enumerated) by 
programmer and represented as identifiers:

(Mon, Tue, Wed, Thu, Fri, Sat, Sun)

An array type is a composite type containing a number of elements of the same type; 
individual elements of an array are accessed by their index (i.e. a position within an 
array), so if a variable v has the type

array [10] of integer

then individual elements of the array v can be accessed as v[0] … v[9] (SPL array 
indexes start with zero, just like in C). Multidimensional arrays can be declared just as 
easily; the number of array dimensions is not limited:

array [10][20][30] of integer

A structure is a collection of values, possibly of different types, whose components 
are accessible by name, so if a variable s has the type

structure
{
    x, y : integer;
    r    : real;
}

then individual elements of s are s.x, s.y and s.r.

A union is a special case of structure where all components overlap in memory, so if a 
variable u has the type

union
{
    i : integer;
    r : real;
}

then it can hold either an integer value (in its component u.i) or a real value (in its 
component u.r), but not both at the same time.

Both structures and unions can be packed. Although packing a structure or union 
may result in less efficient code, packed structures/unions usually occupy less 
memory and their layout in memory is guaranteed to be exactly as specified by the 
programmer – a guarantee C does not provide.

A value of a pointer type points (refers to) a location of some other value in memory. 
For example, a value of a type ^integer points to an integer value. In addition, 
SPL provides a typeless pointer, which points to a location in memory but does not 
specify what type of value is stored there (this is roughly the same as void* in C).

A procedure type describes the signature (i.e. number and types of parameters and 
return value) of a procedure. Each procedure declaration assigns a proper type to the 
procedure being declared; in addition, variables of procedure types can be declared 
and manipulated (roughly equivalent to function pointers in C).

A label type allows treatment of statement labels as first-class values. This allows 
calculated and non-local jumps in situations where normal use of structural 
programming will yield an inelegant solution.

Like C types, SPL types can be modified with constant or volatile keywords, 
signifying, respectively, a value that cannot be changed and a value that can be 
changed by means external to the SPL program.

Unlike C types, SPL types can also be modified with packed keyword, signifying a 
value that does not obey the platform alignment constraints. Use of packed types can 
significantly reduce the amount of memory required for data storage at the expense of 
slower data access and slight increase in the code size.

Another important SPL feature not found in C is uniform type equivalence system. By 
default, all SPL types, including composite types (such as arrays or structures), use 
structural equivalence, i.e. two type signatures consisting of the same sequence of 
symbols always denote the same type. However, SPL allows to explicitly request 
name equivalence when introducing a new type, so the type declaration

type Handle = new integer;

introduces a new type Handle which is similar to integer except it’s a separate 
type – two Handles can be compared to each other, but a Handle cannot be 
compared to or assigned to an integer.

Expressions
SPL expressions are similar to those in C, although some operators in SPL are written 
using keywords instead of special characters. The operators fall into several classes:

Arithmetic: +
-
*
/
%

Logical: and
or
xor
not
implies
and then
or else
implies then

Relational: ==
<>
<
<=
>
>=

Assignment: =

Conditional: ?:

Shift: <<
>>

Increment/decrement: ++
--

Address calculation: &

Dereferencing: ^

Member access: .
->

Dynamic memory allocation: new

Miscellaneous sizeof
old

Most SPL operators are similar to their C counterparts (except C does not provide the 
“implication” operators). Unlike C, SPL allows the programmer to explicitly specify 
whether a logical operation over boolean values shall short-circuit or not (i.e. x 
and y always evaluates both operands, while x and then y only evaluates y if x
is true).

Like C, SPL allows a constant expression anywhere a constant is allowed:

<= Less than or equal

> Greater than

>= Greater than or equal

<< Bit shift left

>> Bit shift right

& Address

++ Increment

-- Decrement

? Conditional

= Assignment

-> Dereferencing

^ Pointer

(  ) Expressions, enumerated types, parameter lists

{  } Blocks

[  ] Array indexing

.  ,  ;  :  ... Punctuation

Comments
Any sequence of characters between the symbols /* and */ is a comment. Such 
comments can appear anywhere within a program, either within a line or extending 
over many lines. Although comments usually appear between declarations or 
statements, they may appear between any two symbols:

constant PI = /* to five decimal places */ 3.14159;

Comments may not divide a symbol, however, so the following example is not a valid 
SPL:

constant Pi = 3.1415/* to five decimal places */9;

Comments cannot nest, so the following example is not a valid SPL either:

constant Pi = /* to five /*decimal places*/ */ 3.14159;

SPL also allows a second form of comment, which starts with the symbol // and 
extends until the end of the line:

constant Pi = 3.14159; // to five decimal places

Usually, the first form is used to write large comments documenting function 
interfaces, etc., while the 2nd form is used to comment individual declarations and 
statements.

additional standard header files), and will, most probably, need to create several 
source files and some header files of our own.

One of the central concepts of SPL programming is that of a compilation unit. An 
SPL compilation unit is a single SPL source file together with all header files included 
into that source file directly or indirectly (indirect inclusion occurs when a header 
files included with the #include preprocessor directives into the source file has 
additional #include directives within, thus causing more header files to be 
included). The name “compilation unit” reflects the use of the concept – a compilation 
unit is what SPL compiler processes atomically, generating one object file. Note that, 
although some SPL compilers may allow compiling several source files in one 
invocation, these source files are compiled independently, each of them being treated 
as a separate compilation unit.

As you can see, SPL does not place any constraints on how the program is split into 
compilation units, how header files are organized, etc. This allows the programmer to 
select the source code organization best suited for the project at hand. Proponents of 
more formal program structure (like that imposed by Module-2 or Ada, where a large 
program is broken into a number of modules or packages, each with a public and 
private part) will be glad to hear that SPL allows emulating the modular program 
organization without any additional effort – all that is required is a little programming 
discipline (for more information on writing modular SPL code see the SPL Coding 
Style Guide).

character*1 ANSI characters.

character*2 Unicode BMP characters.

character*4 ISO-10646 characters.

character Either character*1, character*2 or character*4, 
depending on the platform.

boolean Boolean values (true and false).

Numeric types
All SPL implementations supply the three sets of numeric types: integer types, 
cardinal types and real types. Integer types consist of whole numbers, both positive 
and negative, in a type-specific range. Cardinal types are similar, but include only 0 
and a range of positive integers. Real types represent real numbers, both positive and 
negative. The SPL standard neither requires nor permits any other standard numeric 
types besides those described in the following sections.

Integer types
The range of integer types varies from type to type; values of type integer*N
(where N is 1, 2, 4 or 8) are represented as binary integer values in two’s complement 
format and occupy N bytes of memory. Note, that type names integer*1, 
integer*2, integer*4 and integer*8 are treated as atomic keywords and 
must be written exactly as specified; in particular, no spaces are allowed before or 
after the star character, so integer * 8 (one space before and after the star 
character) is not a valid type name.

SPL defines arithmetic, logical and shift operators for integer types.

Integer arithmetic
The following arithmetic operators are defined for integer types:

Operator Operand types Result type Comment

A + B
A : integer*N
B : integer*N integer*N Result is the sum of operands.

A – B
A : integer*N
B : integer*N

integer*N Result is the difference of 
operands.

A * B
A : integer*N
B : integer*N integer*N Result is the product of 

operands.

A / B
A : integer*N
B : integer*N integer*N

Result is the integer part of 
quotient obtained when A is 
divided by B.

A % B
A : integer*N
B : integer*N integer*N

Result is the remainder from an 
integer division of operands. It is 
always calculated as:
(A%B) = (A-((A/B)*B))

+ A A : integer*N integer*N Result is the same as operand.

- A A : integer*N integer*N Result is the operand negated.

The only operators that deserve special attention are / (“divide”) and % (“modulus”). 
If A and B have type integer*N, then A/B is the quotient obtained when A is 
divided by B (with the fractional portion dropped if the result is not an integer), and 
A%B is the remainder.

Expression Value

6 / 2 3
2 % 2 0
1 / 3 0
1 % 3 1
-3 / 2 -1
-3 % 2 -1
3 / -2 -1
3 % -2 1

Note that the value of A%B is defined precisely for all cases (i.e. A and B can each be 
positive or negative). This is different from many other programming languages, 
where A%B is defined precisely only when A and B are both positive, but is machine-
or implementation-specific if either or both is negative.

The operators *, / and % take precedence over the binary forms of + and -. Binary 
operators at the same level of precedence are evaluated from left to right:

Expression Value

1 – 6 – 4 -9
5 + 8 * 3 29

4 * 10 + 6 / 2 43

The programmer can use parentheses to change the order of evaluation:

Expression Value

1 – (6 – 4) -1
(5 + 8) * 3 39

4 * (10 + 6) / 2 32

Unary operators + and - have higher precedence than any binary operators and are 
evaluated from right to left. SPL does not explicitly forbid adjacent operators:

Expression Value

-9 + 4 / 2 -7
12 + -3 9
12 + (-3) 9

The last two expressions in the above example are actually the same; the difference is 
in that the last one uses explicit parentheses to improve readability.

All integer operators expect their operands to be of the same type and produce the 
result of the same type. This means that when, for example, two integer*1 values 
are multiplied, the result is also an integer*1 value. However, multiplying two 
integer values in range [-128; 127] (which is the integer*1 range of values) can 
actually result in a value outside this range; for example 64 * 64 = 4096. In such 
situation, only the lowest significant bits of the result are available. In the above 
example, 010000002 * 010000002 = 00010000000000002, but only the 
lowest 8 bits of the result are actually kept, so the actual result of the multiplication 
64t * 64t (the suffix t designates integer constant as belonging to type 
integer*1) yields the result 0t. This situation is called an overflow and can occur 
not only when multiplying values, but also when adding or subtracting them. It is 
programmer’s responsibility to detect and, if necessary, handle overflows.

The operators / (“divide”) and % (“modulus”) have yet another special case – that of 
their second operand being 0. Mathematically, a division by zero does not yield a 
finite value – yet both operators must yield the value of the same type as their 
operands. SPL defines both situations to be errors which interrupt the normal 
execution of a program and are handled in an implementation-specific manner. Some 
examples of such handling may be program termination (with or without an error 
message saying “The program is terminated due to an attempt to divide by zero”), 
post-mortem dump or bringing up the debugger.

Bitwise logical operators
The following bitwise logical operators are defined for integer types:

Operator Operand types Result type Comment

A and B
A : integer*N
B : integer*N

integer*N Result is the bitwise logical 
conjunction of operands.

A or B
A : integer*N
B : integer*N integer*N Result is the bitwise logical 

disjunction of operands.

A xor B
A : integer*N
B : integer*N integer*N

Result is the bitwise 
exclusive disjunction of 
operands.

A implies B
A : integer*N
B : integer*N integer*N Result is the bitwise logical 

implication of operands.

not A A : integer*N integer*N Result is the bitwise logical 
negation of operand.

As their name implies, bitwise logical operators applied to integer values are 
performed on all bits of operand(s) independently, yielding corresponding bits of the 
result. The only operator that deserves special attention is implies , which is 
defined as follows: (A implies B) = ((not A) or B).

Bitwise logical operators have the following relative precedence:

Operator Precedence

and Highest
implies
xor
or Lowest

The precedence of all bitwise logical operators is lower than that of arithmetic 
operators. Bitwise logical operators at the same level of precedence are evaluated 
from left to right.

Bitwise integer shifts
The following bitwise shift operators are defined for integer types:

Operator Operand types Result type Comment

A << B
A : integer*N
B : integer*K

integer*N
Result is the operand A
shifted left by B bits.

A >> B
A : integer*N
B : integer*K

integer*N
Result is the operand A
shifted right by B bits.

A << B
A : integer*N
B : cardinal*K

integer*N
Result is the operand A
shifted left by B bits.

A >> B
A : integer*N
B : cardinal*K

integer*N
Result is the operand A
shifted right by B bits.

Bitwise shift operators shift the binary representation of their first operant right (>>) 
or left (<<) by the specified number of positions. Note that, unlike other integer 
operators, shift operators do not require their operands to be of the same type; 
however, their result is always of the same type as their first operand.

When shifting to the left, the new bits introduced by the shift are always 0. When 
shifting right, the new bits introduced by the shift are copies of the highest (sign) bit 
of the first operand.

Note that the number of positions by which to shift (also called the shift counter) can 
be specified as either unsigned (cardinal*K) or signed (integer*K) value. In 
the latter case, if the number of bits to shift is negative the actual shift is performed by 
the specified number of bits in the opposite direction, so (A >> (-3)) is the same 
as (A << 3).

All bits in the shift counter are significant. When the shift counter B in an expression 
A << B (or A >> B) exceeds the actual number of bits in the value being shifted, 
the result of a shift is the same as if the shift was performed by 1 bit and repeated B
times. Therefore:

in order to compare an ANSI character and an ISO-10646 character. Instead, one must 
use an explicit cast:

character*4('a') > i'b'

Note that, in the example above, the smaller character type is casted to the larger 
character type before the two characters are compared. The expression

'a' > character*1(i'b')

would not work in general case, because casting a character*4 value to 
character*1 will lose the 24 most significant bits of the ISO-10646 character 
code, hence the above expression can give a false positive.

The following example program encodes a message as it is typed at the keyboard. It 
uses a simple “rotation” scheme to perform the encoding; it replaces the letters from 
A through M by N through Z, respectively, and replaces the letters from N to Z by A 
through M. Characters that are not letters do not change. The program has two key 
statements. One converts an uppercase letter to its encoded version; the other encodes 
a lowercase letter. The following statement does the uppercase encoding:

ch = character((cardinal(ch) –
               cardinal('A') + 13u) % 26u +
              cardinal('A'));

The idea is to perform a “circular shift” of the letter ch by adding 13 to 
cardinal(ch), taking the result modulo 26, then converting back to character 
form. This operation will work only if the program first subtracts cardinal(‘A‘)
from cardinal(ch) – to scale it into the range between 0 and 25 – and adds it 
after performing the modulus operation. This strategy assumes that uppercase letter 
occupy continuous positions in the character set – which is true for all SPL 
implementations, since they all invariable use the ANSI character set for 
character*1 values.

The following program reads characters typed at the keyboard until the used enters an 
escape character. The program uses the Console::readCharacter,
Console::writeCharacter procedures from the SPL standard library to read 
and write one character at a time; these procedures are declared in the standard header 
file “conio.sph”. Because Console::readCharacter does not echo 
characters typed at the keyboard when console is in raw mode, the uses sees only the 
encoded version of the message, not the original version.

#include "conio.sph"
using spl;

procedure main() public
{
    declare ch : character;

    Console::setMode(Console::Mode::Raw);
    Console::print("Enter the message to be encoded\n");
    Console::print("Indicate end of message by Esc\n");

    Console::readCharacter(ch);
    while (ch <> '\e')
    {
        switch (ch)
        {
            when 'A' ... 'Z':
                ch = character((cardinal(ch) –
                             cardinal('A') + 13u) % 26u +
                            cardinal('A'));
                break;
            when 'a' ... 'z':
                ch = character((cardinal(ch) –
                             cardinal('a') + 13u) % 26u +
                            cardinal('a'));
                break;
        }
        Console::writeCharacter(ch);
        Console::readCharacter(ch);
    }
}

One property of this code is that a second application of the encoding algorithm
undoes the result of the first application. Consequently, the same program can also be 
used for decoding.

The boolean type
The boolean type has only two values, denoted by the keywords true and false. 
The relational operators (equal, not equal, less than and so on) produce boolean
values. The logical operators (and, or, not and so on) require boolean operands 
and produce a boolean result.

Relational operators
SPL provides six relational operators:

== Equal

<> Not equal

< Less than

<= Less than or equal to

> Greater than

>= Greater than or equal to

We can use relational operators to compare values of integer, cardinal, real, character, 
boolean, enumeration or pointer type. SPL considers the boolean value false to 
be less than the value true. Characters are compared using their character codes:

Expression Value

'a' < 'b' true (cardinal('a') is 97 and cardinal('b') is 98)

'a' < 'A' false (cardinal('a') is 97 and cardinal('A') is 65)

In order to compare two values using one of the operators ==, <>, <, <=, > or >=, 
these values must be of the same type.

Logical operators
SPL provides eight basic logical operators: and, and then, or, or else, xor, 
not, implies and implies then. Five of these operators have already been 
mentioned as applicable to integer and cardinal types, where they perform bitwise 
logical operations. These same operators, when applied to boolean operands, yield 
a boolean result. The following table summarizes the logical operators applicable to 
boolean values:

Operator Operand types Result 
type

Comment

A and B
A : boolean
B : boolean boolean Result is the logical 

conjunction of operands.

A and then B
A : boolean
B : boolean

boolean Result is the logical 
conjunction of operands.

A or B
A : boolean
B : boolean boolean Result is the logical 

disjunction of operands.

A or else B
A : boolean
B : boolean boolean Result is the logical 

disjunction of operands.

A xor B
A : boolean
B : boolean boolean Result is the exclusive 

disjunction of operands.

A implies B
A : boolean
B : boolean boolean Result is the logical 

implication of operands.

A implies then B
A : boolean
B : boolean

boolean Result is the logical 
implication of operands.

not A A : boolean boolean Result is the logical negation 
of operand.

The following rules define the values of logical operators:

Operator Has the value true if

A and B Both A and B are true.

A and then B Both A and B are true.

A or B A or B (or both) are true

A or else B A or B (or both) are true

A xor B A or B (but not both) is true.

A implies B A is false or A and B are both true.

A implies then B A is false or A and B are both true.

not A A is false.

In expressions involving two or more logical operators, the following rules of 
precedence apply:

Operator Precedence

not Highest
and, and then

implies, implies then
xor

or, or else Lowest

SPL does not specify the order of evaluation for operators and, implies, or and 
xor. For these operators, the only thing guaranteed is that both operands are always 
evaluated.

The other three logical operators always perform the short-circuit evaluation:

Operator Evaluation rules

A and then B

A is always evaluated first. If A is false, then B is not 
evaluated at all and the operation yields false. 
Otherwise (i.e. when A is true) B is evaluated as well,
and the result of the operation is the same as B.

A or else B

A is always evaluated first. If A is true, then B is not 
evaluated at all and the operation yields true. 
Otherwise (i.e. when A is false) B is evaluated as 
well, and the result of the operation is the same as B.

A implies then B

A is always evaluated first. If A is false, then B is not 
evaluated at all and the operation yields true. 
Otherwise (i.e. when A is true) B is evaluated as well, 
and the result of the operation is the same as B.

Note, that there is no short-circuit form of the “exclusive or” operator, because its 
result always depends on both operands.

Short-circuit evaluation of boolean expressions can be very useful. For example, in 
the expression

(x > 0.0) and then (y / x > 0.0)

the value y / x will not be computed if x is less than or equal to zero, thus 
preventing a division by zero. Programmers often use this kind of expression to 
prevent an array index from going out of range or to test that a pointer is not null
before dereferencing it. For example, if A is an array whose indices range from 0 to N, 
the following loop searches A for the value x:

for (i = 0; i <= N and then A[i] <> x; i++) {}

If x does not appear in A, then i will eventually reach the value N + 1. At this point, 
the expression (i <= N) will be false, so the loop will terminate; the expression 
A[i] <> X will not be evaluated, thus avoiding an index out of range.

Conditional operator
A conditional operator uses special 3-operand syntax to select one of the values based 
on a boolean condition. The general form of a conditional operator is:

<condition> ? <then-value> : <else value>

When a conditional expression is evaluated, the boolean <condition> is always 
evaluated first. Depending on whether the condition if true or false, either 
<then-value> or <else-value> is evaluated and used as the result of the 
conditional operator. Naturally, the two values <then-value> and <else-
value> must be of the same type.

A conditional operator has lower precedence than any other operator (except 
assignment operators, described later in this manual).

Boolean casting
When a numeric value is casted to boolean type, the result is true if the numeric 
value is nonzero and false if it is zero, therefore boolean(1) == 
boolean(2.34), because both casts yield the value true.

When a boolean value is casted to a numeric type, false becomes zero in the 
corresponding numeric type and true becomes 1, so integer(false) == 0
and real(true) == 1.0.

Characters behave similarly to cardinals when used in type casts; therefore all 
characters except the NUL character (character with code 0) become true when 
casted to boolean type, and NUL character becomes false. A cast from a 
boolean type to character yields either a NUL character or a character with the code 
1, depending on whether the boolean value being casted is false or true.

Modified types
SPL allows any type, including the predefined types described above, to be modified. 
Type modification is performed by writing one or more type modifiers before the type 
signature.

SPL supports the following three type modifiers:

Type modifier Meaning

constant

If T is some type, than constant T is a different 
type that has the same properties as T except values of 
the type constant T cannot be assigned to. T and 
constant T are assignment-compatible (i.e. a value 
of type constant T can be assigned to a variable of 
type T).

volatile

If T is some type, than volatile T is a different 
type that has the same properties as T except values of 
the type volatile T can be modified by means 
external to the current thread of the SPL program. T
and volatile T are assignment-compatible (i.e. a 
value of type volatile T can be assigned to a 
variable of type T). The “means external to the current 
thread” can be some other thread of the same program, 
or completely outside program’s control (e.g. a 
memory-mapped I/O address). Therefore, in a 
multithreaded program all static variables used by more 
than one thread must be declared volatile.

packed

If T is some type, than packed T is a different type 
that can store the same set of values as T but does so 
using as little memory as possible. T and packed T
are not assignment-compatible (i.e. a value of type 
packed T cannot be assigned to a variable of type T).

Expressions
Having introduced SPL basic types and their operators, we can now give a complete 
description of the syntax of SPL expressions.

Generally, an expression consists of operands (constants, variables and/or procedure 
calls) joined by operators. Most of the SPL operators have been introduced in 
previous chapters together with SPL types they apply to. In addition, SPL allows 
certain other operators, described below.

Increment and decrement operators
As their name suggests, increment and decrement operators are used to increment or 
decrement a value of a variable. SPL supports 4 such operators:

Operator Meaning

++x Increment the variable x by 1, yielding the new (incremented) 
value of x as the result of an operator.

--x Decrement the variable x by 1, yielding the new (incremented) 
value of x as the result of an operator.

x++ Increment the variable x by 1, yielding the original (before 
increment) value of x as the result of an operator.

x-- Decrement the variable x by 1, yielding the original (before 
increment) value of x as the result of an operator.

Note that increment and decrement operators can only be applied to variables, but not 
constants or sub-expressions, so ++1 and (x+y)-- are both illegal. Only integral, 
real and character values can be incremented/decremented, but not boolean values.

Assignment
The special assignment operator is used when a value of an expression must be 
assigned to a variable. For example, the assignment x = 3 assigns the integer
value 3 to the variable x.

An assignment operator yields a value just like any other operator. The result of an 
assignment operator is the value being assigned. This value can be used to construct 
further expressions, so expression x = y = 3 is valid and assigns an integer
value 3 to variable y and then assigns the same value to variable x. Naturally, both 
variables x and y must be of type integer for an assignment to be valid.

Compound assignment
Compound assignment operators provide a syntactic shortcut for binary operators 
used together with an assignment operator. Any expression of the form x = x 
<binary operator> y can be written instead as x <binary operator> = 
y. The effect of the two forms is the same, except the latter (compound assignment) 
form guarantees that x is evaluated exactly once (and, therefore, all side effects 
caused by evaluation of x occur at most once).

Compound assignment is most useful when a variable must be modified, so x *= 
10 increments the value of x tenfold, whereas x += 1 increments the value of x by 
1 (and is, therefore, equivalent to ++x). Note the absence of space between binary 
operator (+ or *) and assignment operator – the SPL compiler is clever enough to 
realize that += is two distinct operators written one after another.

In more elaborate cases, compound assignment can entail conditional evaluation, e.g. 
an expression b and then = f(x) will perform the procedure call f(x) only if 
b is true and then update b.

The comma operator
The special binary operator , (comma) can be used to join arbitrary expressions. 
When a comma operator is evaluated, the left operand is evaluated first, followed by 
the right operand. The result of a comma operator is the same as the result of its right 
operand.

The sizeof operator
The operator sizeof allows determining the size of any variable, type or value. It is 
written as sizeof(<variable, type or value>) and allows arbitrary 
expressions or arbitrary types as operands. The result of a sizeof operator is always 
a cardinal value representing the number of bytes occupied by the specified value 
or type.

Since it is possible to determine the size of all types and variables at compile-time, 
sizeof can appear in constant expressions (see below) as an operand.

Operator precedence and associativity
SPL operators allow the construction of complicated expressions. To ensure that each 
expression has a unique meaning, SPL provides rules for operator precedence and 
associativity. The following table summarizes these rules:

Operator Associativity Precedence

++(postfix), --(postfix), ^ Left to right Highest

+ (prefix), - (prefix), not, &, 
++ (prefix), -- (prefix)

Right to left

*, /, % Left to right

+, - Left to right

<<, >> Left to right

==, <>, <, <=, >, >= Left to right

and, and then Left to right

implies, implies then Left to right

xor Left to right

or, or else Left to right

?: (conditional) Right to left

=,
+ =, - =, * =, / =, % =,

<< =, >> =,
== =, <> =, < =, <= =, > =, 

>= =,
and =, and then =,

implies =, implies then =,
xor =,

Right to left

or =, or else =

, (comma) Left to right Lowest

The rules for precedence and associativity come into play whenever an expression 
contains more than one operator. If the expression contains operators at different 
levels of precedence, the program performs the operations in the order of their 
precedence. If several operators are at the same precedence, the associativity rule 
dictates the order in which these operators are performed.

For example:

Expression Order of evaluation

i * 3 + j / 2 (i * 3) + (j / 2)
i / j * k (i / j) * k

- j – j - k ((-i) – j) - k
p or q and r p or (q and r)
p and q and r (p and q) and r

p and not q or r (p and (not q)) or r
1 + 2 * j < k + 1 (1 + (2 * j)) < (k + 1)

Note that the order in which operands of binary operators are evaluated is not fixed 
(with the exception of short-circuit boolean operators and then, implies then
and or else).

We can use parentheses to change the order of evaluation within an expression. When 
in doubt, and particularly when expressions are complicated, use parentheses 
liberally.

Note that, unlike many other languages, SPL allows any number of prefix or postfix 
operators to be specified in a sequence, so the expression - - - a is valid and 
yields the same result as – a. Note, however, that spaces are still mandatory as 
operator separators; in the absence of space the expression --a will be interpreted as 
pre-decrement, whereas - - a changes the sign of a twice (thus always yielding a
itself).

Constant expressions
In SPL, a constant expression may appear wherever a language allows a constant. A 
constant expression has the same syntax as an ordinary expression. However, a 
constant expression must contain only constants, not variables or function calls, as 
operands. The following are examples of constant expressions:

13.5 * 1.0567 / 4.0
(x <> 0) ? (y / x) : (y)
('+' < '-') or else ('Z' >= 'z')

The only identifiers that can appear in constant expressions are identifiers bound to 
constant values by means of constant declarations (described in the next chapter).

Constant and variable declarations
SPL provides two ways to associate an identifier with a data value: by a constant 
declaration or by a variable declaration. A constant declaration specifies an identifier 
and a constant value; the association between the identifier and the value is permanent 
and cannot be changed by assignment. A variable declaration specifies an identifier 
and a type. No fixed value is associated with the identifier; its value changes during 
program execution as the result of assignment.

Constant declarations
A constant declaration binds an identifier to a constant value; each subsequent 
appearance of the identifier is equivalent to an appearance of the constant value that it 
represents. A constant declaration has the following general form:

constant <identifier> = <value>,
         . . .
         <identifier> = <value>;

Here are several constant declarations:

constant SpeedOfLight      = 1.86E5,
         AbsoluteZero      = -273.1,
         MonthlySalesGoal = 1000,
         QuestionMark      = '?',
         LineFeed         = '\n',
         DebugOption       = true,
         ErrorMessage      = "END expected";

As the example shows, constant can represent values of any type, including numeric, 
character, boolean and string. SPL naming conventions suggest constant are given 
PascalCased names, in order to make them stand out in the program text.

Arbitrary constant expressions may appear to the right of the equal sign:

constant Pi     = 3.14159,
         Radius = 10.0,
         Area   = Pi * Radius * Radius,
         First = 0,
         Last   = 99,
         Length = Last – First + 1;

The programmer does not have to declare a constant before using it to declare other 
constants. In the example above, the declaration of Area relies on the previously 
declared constants Pi and Radius. An attempt to declare Area before Pi and 
Radius would, however, still be legal:

constant Area   = Pi * Radius * Radius, // OK
         Pi     = 3.14159,
         Radius = 10.0;

procedure main() public
{
    constant Pi         = 3.14159,
             TwoTimesPi = 2.0 * Pi;

    declare radius : real;

    Stdout::print("Enter radius:");
    Stdin::scan("{real}", &radius);
    Stdout::print("\n");

    Stdout::print("Circumference is: {real}\n",
                  TwoTimesPi * radius);
    Stdout::print("Area is: {real}\n",
                  Pi * radius * radius);
}

As any other named entities, variables do not have to be declared before use; therefore 
the above example could be rewritten as:

#include "io.sph"
using spl;

procedure main() public
{
    Stdout::print("Enter radius:");
    Stdin::scan("{real}", &radius);
    Stdout::print("\n");

    Stdout::print("Circumference is: {real}\n",
                  TwoTimesPi * radius);
    Stdout::print("Area is: {real}\n",
                  Pi * radius * radius);

    declare radius : real;

    constant TwoTimesPi is 2.0 * Pi,
             Pi         is 3.14159;
}

This version, where constants and variables are declared after their use, is still a valid 
SPL. However, declaring variables (and other program entities) after use makes 
reading programs more difficult and is, therefore, not generally recommended.

Statements
This chapter presents eleven of SPL’s fifteen statements. Apart from an expression 
statement, these represent control structures, which control the order in which 
statements are executed.

SPL statements can be divided into three groups. Statements in the first group are for 
selection; choosing between two or more computation paths. The other control 
structures provide ways to perform iteration or looping – the repeated execution of a 
sequence of statements. Statements in the third group perform various actions.

The following table summarizes SPL control structures:

Statement Use Effect

expression calculation Evaluate an expression. Side effects can 
include procedure calls, assignments, etc.

empty - No effect

block grouping Treat a sequence of statements as a single 
statement

if selection Choose among alternatives based on a 
boolean expression

switch selection Choose among alternatives based on a value 
of an expression

while iteration Repeat as long as a boolean expression 
remains true

do iteration Repeat as long as a boolean expression 
remains true

for iteration Repeat as long as a boolean expression 
remains true

goto control transfer Transfer control to the designated statement

break control transfer Exit loop or selection statement

continue control transfer Continue loop iteration

The expression statement
The SPL expression statement has the form of an arbitrary expression followed by a 
semicolon. When an expression statement is executed, the expression is evaluated.

An expression statement will typically contain at least one side effect, such as
assignment of a value to a variable, a procedure call, etc. Expression statements 
without side effects, although permitted by the language, do not have any effect when 
executed.

{
    result = dividend / divisor;
    remainder = dividend % divisor;
}
else
{
    result = 0;
    remainder = 0;
}

programmers can omit the else clause if no action is to be taken when the condition 
is false:

if (divisor = 0)
{
    divisor = 1;
}
result = dividend / divisor;

Programmers can nest if statements. For example, the following if statement will 
find the largest of the three numbers num1, num2 and num3:

if (num1 < num2)
    if (num2 < num3)
        largest = num3;
    else
        largest = num2;
else
    if (num1 < num3)
        largest = num3;
    else
        largest = num1;

As the above example shows, nested if statements can be confusing to the reader. 
Therefore, it is recommended to use block statements in order to avoid confusion. The 
above example can be rewritten as:

if (num1 < num2)
{
    if (num2 < num3)
        largest = num3;
    else
        largest = num2;
}
else
{
    if (num1 < num3)
        largest = num3;
    else
        largest = num1;

}

Allowing if statements to nest may leads to a dangling-else problem, as in the 
following example:

if (x > 0)
    if (y > 0)
        result = x + y;
    else
        result = x - y;

In such situation, the general rule is that else is related to the closest preceding if
for which there has not yet been an else. Therefore, the above example is equivalent 
to:

if (x > 0)
{
    if (y > 0)
        result = x + y;
    else
        result = x - y;
}

To force else to relate to the first if statement instead, a block statement must be 
used:

if (x > 0)
{
    if (y > 0)
        result = x + y;
}
else
    result = x - y;

The switch statement
Consider a problem of translating a letter on a telephone dial or key to the 
corresponding digit. Assume that the variable letter stores a capital letter other 
than Q or Z (the letters Q and Z don’t appear on a telephone dial). We must map the 
letter to one of eight digits, so we could use an if statement controlled by eight 
boolean expressions:

if (letter == 'A' or else 
    letter == 'B' or else 
    letter == 'C')
{
    digit = 2;
}
else if (letter == 'D' or else 
         letter == 'E' or else 
         letter == 'F')

{
    digit = 3;
}
else if (letter == 'G' or else 
         letter == 'H' or else 
         letter == 'I')
{
    digit = 4;
}
else if (letter == 'J' or else 
         letter == 'K' or else 
         letter == 'L')
{
    digit = 5;
}
else if (letter == 'M' or else 
         letter == 'N' or else 
         letter == 'O')
{
    digit = 6;
}
else if (letter == 'P' or else 
    letter == 'R' or else 
    letter == 'S')
{
    digit = 7;
}
else if (letter == 'T' or else 
         letter == 'U' or else 
         letter == 'V')
{
    digit = 8;
}
else if (letter == 'W' or else 
         letter == 'X' or else 
         letter == 'Y')
{
    digit = 9;
}

We can replace this rather lengthy if statement with a much simpler switch
statement:

switch (letter)
{
    when 'A': when 'B': when 'C':
        digit = 2;
        break;
    when 'D': when 'E': when 'F':
        digit = 3;
        break;

    when 'G': when 'H': when 'I':
        digit = 4;
        break;
    when 'J': when 'K': when 'L':
       digit = 5;
        break;
    when 'M': when 'N': when 'O':
        digit = 6;
        break;
    when 'P': when 'R': when 'S':
        digit = 7;
        break;
    when 'T': when 'U': when 'V':
        digit = 8;
        break;
    when 'W': when 'X': when 'Y':
        digit = 9;
        break;
}

A switch statement contains a switch selector (enclosed in parentheses after the 
keyword switch) and a series of cases. Each case consists of a series of values (case 
labels) followed by one or more statements. When a switch statement is executed, 
the value of the switch selector is evaluated and compared to the case labels. If the 
value matches one of the case labels, execution continues from the point where this 
case label occurs, until the end of switch statement is reached or until a break
statement is encountered.

Note the use of break statements in the above example to exit the switch
statement after the digit has been assigned a value. In the absence of break
statements, control will “fall through” the cases, causing digit to always be 
assigned the value 9.

If it is possible that the value of the switch selector might not match any of the case 
labels, a switch statement can contain a default case. In our example, if letter
might have a value other than ones listed, the switch statement shall have a default
case:

switch (letter)
{
    when 'A': when 'B': when 'C':
        digit = 2;
        break;
    when 'D': when 'E': when 'F':
        digit = 3;
        break;
    when 'G': when 'H': when 'I':
        digit = 4;
        break;
    when 'J': when 'K': when 'L':

        digit = 5;
        break;
    when 'M': when 'N': when 'O':
        digit = 6;
        break;
    when 'P': when 'R': when 'S':
        digit = 7;
        break;
    when 'T': when 'U': when 'V':
        digit = 8;
        break;
    when 'W': when 'X': when 'Y':
        digit = 9;
        break;
    default:
        Stdout::print("Error: invalid letter\n");
        break;
}

The default case, if specified, does not necessarily have to be the last one.

If the switch selector value does not match any of the case labels and the switch
statement does not have a default case, the switch statement has no effect.

Case labels are most often just constants, but can be constant expressions (remember 
SPL’s general rule: a constant expression may appear wherever the language allows a 
constant). Programmers can abbreviate a sequence of consecutive case labels by 
writing the first and last label, separated by the three consecutive dots:

switch (letter)
{
    when 'A' ... 'C':
        digit = 2;
        break;
    when 'D' ... 'F':
        digit = 3;
        break;
    when 'G' ... 'I':
        digit = 4;
        break;
    when 'J' ... 'L':
        digit = 5;
        break;
    when 'M' ... 'O':
       digit = 6;
        break;
    when 'P': when 'R' ... 'S':
        digit = 7;
        break;
    when 'T' ... 'V':
        digit = 8;

        break;
    when 'W' ... 'Y':
        digit = 9;
        break;
}

No case label may appear more than once in a switch statement; similarly, if case 
ranges are specified they cannot intersect. The type of each label must be the same as 
the type of the switch selector. Only primitive discrete values can be used as switch 
selectors; these include integral (integer and cardinal), boolean, character, and 
enumerated values.

In theory, switch statements are unnecessary, because we can translate every 
switch statement into an equivalent if statement. In practice, however, a switch
statement is usually much simpler than the corresponding if statement. Furthermore, 
the switch statement is often more efficient, particularly if the number of case 
labels is large.

The while, do and for statements
The while, do and for statements are closely related; they all provide a means of 
indefinite iteration. The first two allow a sequence of statements (the body of the 
loop) to execute repeatedly under the control of a boolean expression, while the 
for statement is better suited for loops with a loop counter.

The while statement
The while statement allows a statement or a sequence of statements to execute 
repeatedly as long as a stated condition remains true. For example, suppose that we 
want to read a series of integers and find their sum, stopping when we read a negative 
value. Using a while statement, we can write:

declare i, sum : integer;

sum = 0;
Stdin::scan("{integer}", &i);
while (i >= 0)
{
    sum += i;
    Stdin::scan("{integer}", &i);
}

When the while statement is first reached, the expression i >= 0 is evaluated. If 
its value is true, then the body of the loop, the body of the loop (the statements sum 
+= i; and Stdin::scan("{integer}", &i);) is executed and the 
expression is evaluated again. If the expression is again true, then the body is 
executed once more. The process continues until the value of the expression is 
false. At this point the loop terminated and the statement following the while
statement will be executed next.

The most important fact to remember about the while statement is that its 
termination condition is tested before the body of the loop is executed. If the 
condition is false when control reaches the while statement, then the body is not 
executed at all. Note also that, if the condition is true initially, the while statement 
will not terminate unless the statements in the loop body change the value of the 
condition.

In the above example, the body of a while statement is a block statement. SPL 
allows the body of a while statement (indeed, a body of any loop) to be any 
statement at all; therefore, the above example can be rewritten as:

declare i, sum : integer;

sum = 0;
Stdin::scan("{integer}", &i);
while (i >= 0)
    sum += i, Stdin::scan("{integer}", &i);

The body of the while statement is now a single expression statement, with the 
comma operator (,)  ensuring the proper sequence of actions.

Note that, just as for selection statements, the brackets around the condition are part of 
the SPL syntax and cannot be removed, so writing while i >= 0 instead of 
while (i >= 0) is an error.

The do statement
The do statement is similar to the while statement, except that the body of the loop 
is executed before a condition is evaluated. Therefore, the body of a do statement is 
always executed at least once.

The do statement is useful when the first execution of the loop body establish values 
that the termination condition is to test. For example, consider the problem of 
prompting a user to enter a value within a particular range (between 1 and 10, say). If 
the value typed by the user is invalid, the program should repeatedly prompt the user 
to enter valid input. Using the while statement we might write:

declare number : integer;

Stdout::print("Enter a number between 1 and 10: ");
Stdin::scan("{integer}", &number);
while (number < 1 or else number > 10)
{
    Stdout::print("Enter a number between 1 and 10: ");
    Stdin::scan("{integer}", &number);
}

Unfortunately, the two statements in the body of the while loop also appear before 
the while statement to give number an initial value. We could eliminate them by 





65

i = 0;
while (i < 100)
{
    data[i] = 0;
    i++;
}

(Note that indexes are varied from 0 to 99; this is necessary because in SPL all array 
indexes start from 0).

Using the for statement, the same task can be achieved as:

declare data : array [100] of integer,
        i    : integer;

for (i = 0; i < 100; i++)
{
    data[i] = 0;
}

To simplify things even further, SPL allows declaring the loop counter (i in the 
example) directly in the for statement header:

declare data : array [100] of integer;

for (declare i : integer = 0; i < 100; i++)
{
    data[i] = 0;
}

Apart from smaller code, declaring the loop counter in the for statement limits the 
scope of the loop counter to the for statement itself; in the above example the 
variable i cannot be used outside of the for loop because it is not declared there.

Incidentally, SPL allows declaring variables in all selection and iteration statements
(except for the do loop, which has no “statement header” as such), so the following 
examples are all legal:

while (declare x : integer = y / z + c)
{
    Stdout::print("{integer}\n", x);
    …
}

if ((declare divisor : real = x + y – z) == 0.0)
{
    Stdout::print("Division by zero!\n");
}

else
{
    Stdout::print("Result = {real}\n", dividend / 
divisor);
}

switch (declare ch : character = getNextCharacter())
{
    when 'a' ... 'z':
    when 'A' ... 'Z':
        Stdout::print("Letter: {character}\n", ch);
        break;
    when '0' ... '9':
        Stdout::print("Digit: {character}\n", ch);
        break;
}

Note the extra brackets in the 2nd example – these are necessary to ensure that 
assignment of value to divisor is performed before comparing this value to 0.0.

The break statement
The break statement is only allowed in the body of a switch, while, do or for
statement. When it is executed, the innermost switch, while, do or for statement 
is terminated, and the execution continues from the statement immediately following 
the switch, while, do or for statement.

Recall the example that we used earlier to illustrate the while statement:

declare i, sum : integer;

sum = 0;
Stdin::scan("{integer}", &i);
while (i >= 0)
{
    sum += i;
    Stdin::scan("{integer}", &i);
}

The statement Stdin::scan("{integer}", &i); appears twice: once before 
the loop (to read the first value of i) and once inside the loop (to read the next value 
of i). This duplication is typical of while statements, because the test for 
termination must come at the beginning of the loop. By using a for statement and 
moving the test for termination to the middle of the loop, we can eliminate the first 
call to Stdin::scan:

declare i, sum : integer;

sum = 0;

for (; ; )
{
    Stdin::scan("{integer}", &i);
    if (i < 0) break;
    sum += i;
}

In this example, we gain little by using for statement instead of a while statement. 
But if we can save several statements in this fashion, the program becomes easier to 
read and modify.

If statements often control break statements:

if (i < 0) break;

When we use an if statement in this fashion, we can write the entire statement on a 
single line instead of two lines; this construct is so common that it is worth masking 
an exception to our usual style.

The break statement described so far can exit only one innermost switch, while, 
do or for statement. If it becomes necessary to exit several nested switch, while, 
do or for statements at the same time, the break statement shall specify the label 
of the switch, while, do or for statement to exit.

For example, consider initializing the 3-dimensional array of integers with values 
typed by the user. If any value is negative, the initialization shall stop at once.

declare data : array [2][3][4] of integer;

InitLoop:
for (declare i : integer = 0; i < 2; i++)
{
    for (declare j : integer = 0; j < 3; j++)
    {
        for (declare k : integer = 0; k < 4; k++)
        {
            Stdin::scan("{integer}", &data[i][j][k]);
            if (data[i][j][k] < 0) break InitLoop;
        }
    }
}

Note the use of the label in the break statement. The label InitLoop is associated 
with the outer loop, so the statement break InitLoop; exits that outermost loop 
(as well as both nested loops). In the absence of the label, the statement break;
would have only exited the innermost loop. Note also the use of locally declared loop 
counters, which helps to avoid polluting the outer scope with variables that have no 
business being there.

The words in out that appear in the declaration of parameters x and y indicate that 
both x and y are passed both into and out of the procedure, so that the caller will 
actually see the effect of swapping. swap also contains a declaration of a variable 
called temp that the procedure uses for temporary storage during the exchange of x
and y.

Both fatalError and swap are subroutines, not functions. Functions are discussed 
later on in this section; however, much information in this chapter pertains to 
functions as well as to subroutines.

A procedure declaration has three parts: (1) the procedure heading, which gives 
information necessary to call the procedure (2) the procedure contract, which specifies 
how the procedure behaves from the caller’s point of view and (3) the procedure 
body. A procedure body, which must be a block, contains declarations of entities 
belonging to the procedure and statements to be executed when the procedure is 
called..

Procedure calls
A procedure all consists of a procedure name and a list of actual call arguments 
enclosed in brackets. The call of a subroutine (i.e. a procedure which does not return a 
value) is always a statement (functions can be called in expressions, as described 
below). For example, a call of Swap looks like this:

swap(lower, upper);

When program control reaches this statement, Swap begins to execute. When the last 
statement in the body of Swap has been executed, the procedure returns; the next 
statement to be executed is the statement after the call of Swap.

A call of a procedure without parameters consists of the procedure name followed by 
a pair of brackets with nothing inside them:

fatalError();

Note that in this case brackets are mandatory even though there are no arguments 
enclosed within.

In general case, the programmer is not required to use an identifier to specify the 
procedure being called. Instead, an arbitrary expression specifying the call target can 
be used. The reason for allowing this is that SPL allows programs to treat procedures 
(i.e. both functions and subroutines) as data values; for example, a program can store 
a procedure in an array or structure and later call it.

Declaration of local entities
A procedure body is an arbitrary block statement. As such, it can itself include 
declarations as well as statements. Typically these are declarations of constants and 
variables, but types and procedures can be declared locally as well. All entities 
declared within a procedure are local to the procedure; they exist only when the 

procedure is being executed and their names are visible only inside the block where 
they are declared.

Storage for procedure’s local variables is allocated (on a run-time stack) when the 
procedure is called. When the procedure returns, this storage is released, thereby 
becoming available for subsequent procedure calls. Consequently, the values of local 
variables are lost between calls of the procedure. If these values must be kept, local 
variables can be declared to be static. The storage for local static variables is allocated 
once when the program starts and remains allocated until the program ends. 
Consequently, local static variables retain their values between procedure calls.

The return statement
Normally, a subroutine (i.e. a procedure which does not return a value) returns when 
the last statement in its body has been executed. A procedure may also choose to 
return prematurely by executing a return statement. For example, the following 
procedure searches a line of input for a particular character; when it finds the 
character, the procedure immediately returns:

procedure searchForChar(ch : character)
{
    /* Read characters until ch is found or the end
       of line is reached, issue an error message
       in the latter case */
    declare inputChar : character;

    do
    {
        Stdin::readCharacter(inputChar);
        if (inputChar == ch) return;
    } while (inputChar <> ‘\n’);
    Stdout::print("Failed: character not found\n");
}

If a return statement is the only statement inside an if statement, we usually write 
the if statement on a single line; we adopted the similar convention for break and 
continue statements.

In a function, the return statement serves a second purpose: in addition to causing 
the function to return, a return statement indicates the value that the function 
should return. In a function, an expression (the value to return) must follow the 
keyword return; in a subroutine an expression must not follow the keyword 
return. It is an error for a function to “fall through” without executing a return
statement, while, in a subroutine, using return statement is optional.

Order of procedure declarations
As is the general rule in SPL, procedures can be declared in any order; thus a program 
may call a procedure before its declaration has appeared.

the parameter passing mode, as we will see later). A call of SolveQuadratic
might look like this:

solveQuadratic(1.0, -8.0, 15.0, r1, r2);

The call arguments are the expressions 1.0, -8.0 and 15.0 and the variables r1
and r2; they match the formal parameters a, b, c, root1 and root2 in that order.

SPL provides three parameter passing modes: in, out and in out. These modes 
specify how the parameter behaves during the call of the procedure. Parameters 
declared as in are passed from the caller into the called procedure; any modification 
the procedure makes to these parameters does not affect the caller. Parameters 
declared as out are passed in the other direction; no value is passed from the caller to 
the called procedure, instead the procedure can assign values to these parameters and 
the caller will see the effect of the assignment when the procedure returns. The in 
out is a combination of the two above modes; a parameter declared as in out is 
passed from the caller to the called procedure and then back to the caller upon 
procedure return; any modifications made by the called procedure to the value of an 
in out parameter are visible to the caller.

In a procedure heading, the keyword(s) designating parameter passing modes can be 
written after the colon separating parameter names and parameter properties. If the 
parameter passing mode is not explicitly specified for some parameters, in is 
assumed by default.

In the procedure solveQuadratic, we made a, b and c in parameters, because 
the procedure needs the values of these parameters. We could have made them in 
out parameters because solveQuadratic does not change their values. Making 
them in parameters is preferable, because doing so documents the fact that 
solveQuadratic does not modify their values (incidentally, in parameters result 
in faster code than out or in out parameters for primitive data types such as 
real).

We have declared root1 and root2 to be out parameters. however. By assigning 
values to root1 and root2, solveQuadratic can change the values of the 
corresponding call arguments; these values are then available when the procedure 
returns. For example, after the call of the solveQuadratic shown earlier, the 
values of r1 and r2 are 5 and 3, respectively. Declaring root1 and root2 to be in 
parameters would be a mistake, because assignments to root1 and root2 in the 
procedure body would not affect r1 and r2. Declaring them as in out parameters 
could do the trick, but would also require the caller to assign values to both r1 and 
r2 before calling solveQuadratic (because these values must then be passed into 
the solveQuadratic), thus making the procedure inconvenient to use.

When declaring a procedure with no parameters, the programmer must still write the 
brackets after the procedure name, even though there is nothing inside:

procedure p() ...

in parameters
The call argument corresponding to an in parameter can be any kind of expression 
(including, but not limited to, constants and variables). The type of the call argument 
must be assignment-compatible with the type of the in parameter. The assignment 
compatibility requirement allows the types of the call argument and formal parameter 
to differ in some cases (in particular, a formal parameter may be a pointer while the 
corresponding call argument is an array).

When a procedure with an in parameter is called, the corresponding call argument is 
evaluated and its value copied to the formal parameter, which then serves as a local 
variable within the procedure. The procedure can change the value of the formal 
parameter by assignment, but any change will not affect the call argument, even if it is 
a variable. For example, suppose that we write the procedure WriteLines, which 
writes the specified number of blank lines to the standard output, as follows:

procedure writeLines(n : cardinal)
{
    while (n > 0u)
    {
        Stdout::print("\n");
        n--;
    }
}

writeLines changes the value of n by decrementing it each time through the body 
of the while statement. But because n is an in parameter, only the procedure’s local 
copy changes; the call argument does not. An in parameter behaves exactly like a 
local variable; the only difference between the two is than an in parameter has an 
initial value (the value of the matching call argument).

So that changes to an in parameter will not affect the corresponding call arguments, 
an SPL system must copy the call argument to the in parameter each time the 
procedure is called. If the call argument requires a large amount of storage (for 
example, if it is a large array), this copying can be expensive in terms of both time 
and space. in out parameters (described below) incur no such penalty.

out parameters
The call argument corresponding to an out parameter must be a variable, an array 
element, a structure field or an anonymous variable accessed through the pointer. It 
cannot be a constant or any other kind of expression, because out parameters are 
assigned values in the procedure bodies.

Unlike in parameters, which require that the types of the parameter and the 
corresponding call argument be assignment-compatible, out parameter and the 
corresponding call argument must have the same type (the only exception to this rule 
is related to array out parameters and is described in the section dedicated to arrays).

When a procedure with an out parameter is called, the address of the corresponding 
call argument is evaluated and passed into the procedure. Any access from within the 
procedure to the out parameter uses this address (i.e. out parameters are always 
passed by reference), which remains unchanged for the duration of the procedure call.

The compiler assumes that out parameters are not assigned values before the 
procedure call. Therefore, it will complain about an attempt to use out parameters in 
the procedure body before they are explicitly assigned values.

in out parameters
An in out parameter is an extended form of out parameter, which behaves exactly 
like the out parameter but assumes that the corresponding call argument has been 
assigned a value before the procedure call.

Like out parameters, in out parameters are always passed by reference. Unlike 
out parameters (and like in parameters), in out parameters have initial values 
when the procedure body begins to execute.

A special case of in out parameters constitutes in out parameters of constant
types. Like all in out parameters, these constant in out parameters are 
passed by reference. However, the fact that the parameter type has constant
modifier prohibits assigning new values to these parameters within the procedure 
body. As a consequence, the call argument corresponding to the constant in 
out parameter can be an arbitrary expression. When a procedure with a constant 
in out parameter is called, the value of the corresponding call argument is 
evaluated and stored in a temporary variable (unless it’s already stored in some 
variable), and the address of this temporary variable is then passed into the procedure.

Due to their nature, in out parameters are the most general parameter passing 
mechanism provided by SPL. In particular, in out parameters can be used for 
efficiently passing large data structures in and out of procedures; whether the data 
structure can be modified from within the procedure depends on whether the in out
parameter is of a constant type or not.

Functions
Every SPL procedure is either a subroutine or a function. A function returns a single 
value; furthermore, calls of function procedures can appear in expressions. For 
example, the expression on the right side of the following assignment contains calls of 
the sin and cos functions (incidentally, both spl::sin and spl::cos are 
standard functions provided by the SPL standard library):

tanx = sin(x) / cos(x);

A call of a subroutine, however, is almost always a statement (exceptions to this rule 
are situations where expression operands do not require values – such as first operand 
of a comma expression, etc., so if subr() is a subroutine than expression 
(subr(), x) is legal because the fact that subr() does not return a value does 
not really matter, as this value would have been ignored anyway. Similarly, is 

subr1() and subr2() are two subroutines, then x?subr1():subr2() is a 
valid expression which yields no value).

A function declaration differs only slightly from the declaration of a subroutine. First, 
the function’s heading must specify the type of value it returns. Second, the body of 
the function must include the return statement that, when executed, indicated the 
value that the function returns.

The following declaration of the squareRoot function illustrates these differences:

function squareRoot(n : cardinal) : cardinal
{
    declare root : cardinal;

    root = 1u;
    while (root * root <= n)
    {
        root++;
    }
    return root – 1u;
}

The appearance of the colon following squareRoot’s parameter list indicates that 
SquareRoot is a function rather than a subroutine; the type name cardinal after 
the colon specifies that squareRoot returns a value of type cardinal. The 
return statement indicates that squareRoot returns the final value of the 
expression root – 1u.

Any expression can appear after the keyword return, provided that its type is 
assignment-compatible with the type specified in the heading of the function 
(cardinal, in this case). The body of the function may contain more than one 
return statement; the first one encountered during the execution of the function 
causes it to return. Reaching the end of a function body without executing a return
statement is an error.

Like a subroutine call, a function call consists of a function designator followed by a 
list of call arguments enclosed in parentheses:

squareRoor(a * a + b * b)

A function call may appear wherever a value of the corresponding type can. Some 
examples:

declare i, j, sum : cardinal;

sum = squareRoot(i) + squareRoot(j);
Stdout::print("{cardinal}\n", squareRoot(i + j));

Function can return values of any type, including predefined and modified types.

    randomCardinal = ((Multiplier *
                       randomCardinal) +
                      Increment) % Modulus;
}

In a call of a function with no parameters, just like in a call of subroutine with no 
parameters, an empty list of call arguments must follow the name of the function:

rand = random();

SPL requires the empty parentheses because the language allows procedures to be 
treated as data (this feature is described later in this manual). An SPL compiler would 
interpret the statement:

rand = random;

as an assignment of the random function itself to the variable rand.

Scope and visibility
When a program contains procedures, the same identifier may appear in different 
places in the program representing different entities. We need rules to determine the 
meaning of each occurrence of an identifier, as the following example illustrates:

procedure outer()
{
    declare x : integer;

    procedure inner()
    {
        declare x : integer;
        x = 0;
    }
}

Does the assignment to x in the body of inner refer to the variable declared in 
outer, or to the one declared in inner?

The problem in determining the meaning of an identifier is complicated by the fact 
that a procedure declaration may contain other procedure declarations. (In theory, this 
nesting can be arbitrarily deep; in practice, programmers rarely nest procedures more 
than two or three deep).

Determining the scope of an entity
Each entity (constant, variable, type, procedure or label) in an SPL program has a 
scope – the region of the program in which references to the entity may appear; we 
say that the entity is visible in this region. The scope of an entity x declared in a 
procedure outer if the entirety of outer, with the possible exception of procedures 
or blocks nested into outer. More formally, the following rules apply:

Counting words in a text file
To show how nested procedures arise naturally from top-down design, let’s write a 
program countWords to count the number of words and sentences in a file of 
English text.

First we notice that countWords has two main tasks – counting the words and 
counting the sentences (while reading the input file) and printing the statistics thereby 
accumulated – so we decide to write two procedures collectStatistics and 
printStatistics. The body of countWords will open the input file, call 
collectStatistics and printStatistics, then close the input file.

printStatistics consists of calls of output procedures, so it needs no further 
decomposition. We next decide on a strategy for collectStatistics: it will 
read characters one at a time from the input file, incrementing the appropriate counter 
when it encounters a word or sentence boundary. We delegate these subtasks –
detecting a word boundary and detecting a sentence boundary – to two procedures 
incWordCount and incSentenceCount. Because these procedures are needed 
by collectStatistics but not by printStatistics, we nest them inside
the declaration of the former procedure.

incWordCount can detect the beginning of a word by looking for a whitespace 
character followed by a non-whitespace character. (a whitespace character is a 
character that appears as white space when printed, such as blank, tab, line feed and 
carriage return.) incSentenceCount can detect the end of a sentence by looking 
for a period, question mark or an exclamation mark, followed by a whitespace 
character or right parenthesis.

Both incWordCount and incSentenceCount need to be able to detect 
whitespace characters. Fortunately, the SPL standard library provides a function that 
does just that, so we don’t need to program it ourselves.

The following figure shows the structure of countWords:

Unlike out example programs in previous chapters, countWords must read from a 
file. One advantage of the SPL standard library is that reading from a file is just as 
easy as reading from a terminal. A program uses similar procedures to read from a 

terminal or file; the only difference is that the program must open the file before 
reading it and close the file when it is no longer needed.

#include "io.sph"
#include "fcntl.sph"
#include "ctype.sph"
#include "process.sph"
using spl;

procedure countWords()
{
    declare inputFileName : array [MAX_PATH_LENGTH] 
                            of character,
            inputFile     : File,
            numWords      : cardinal,
            numSentences  : cardinal;

    Stdout::print("Enter the name of the input file: ");
    Stdin::scan("{string|%.*s}", inputFileName,
                                 MaxPathLength – 1u);
    inputFile = File::open(inputFileName,
                           File::OpenMode::OpenExisting,
                           File::AccessMode::Read,
                           File::ShareMode::Read);
    if (inputFile == File::None)
    {
        Stdout::print("Cannot read the file\n");
        Process::exit(Process::ExitCode::Failure);
    }
    collectStatistics(inputFile, numWords, numSentences);
    closeFile(inputFile);
    printStatistics(numWords, numSentences);

    procedure collectStatistics(
            inputFile    : File,
            wordCount    : out cardinal,
            sentenceCount : out cardinal
        )
    {
        declare previousChar : character = ' ',
                currentChar  : character;

        wordCount = sentenceCount = 0u;
        while (File::readCharacter(inputFile,
                                   currentChar))
        {
            incWordCount();
            incSentenceCount();
            previousChar = currentChar;
        }

accesses them (allowing wordCount and sentenceCount to be in
parameters in printStatistics). We could have written 
collectStatistics and printStatistics without parameters; they 
would instead communicate directly through numWords and 
numSentences.

Programmer-defined types
So far only basic SPL types have been described. This chapter and the next three 
show how to augment this collection by defining new types.

Some SPL types are predefined; an SPL compiler recognizes these standard types 
without the programmer having to define them. The basic types described in previous 
chapters are all standard; the only other standard types are pointer and label
described in subsequent chapters. In addition to using standard types, SPL 
programmers can define new types.

Every SPL type, whether standard or not, falls into one of five categories: it is either a 
primitive, composite, pointer, procedure or label type.

A value of a primitive type is a single, indivisible item of data, such as a number or 
character. All basic types described so far are primitive types. SPL also allows 
programmers to create new primitive types in two ways: (1) by listing (enumerating) 
the values of the new type or (2) by taking an existing primitive type and deriving a 
new primitive type from it. We call these enumerated types and derived types, 
respectively.

A value of a composite type is a collection of data items. The programmer must 
define all composite types. Composite types come in three varieties: array types, 
structure types and union types.. When creating a new composite type, the 
programmer must specify how many components a value of the type will contain, 
what the types of these components will be, and how to access the components. 
Composite types are described in the separate chapter later on in this manual.

Pointer, procedure and label types are special-purpose types that share few properties 
of the other types. A value of a pointer type is a reference to another data value; a 
value of a procedure type is a reference to a procedure and a value of a label type is a 
reference to an executable statement. These types are described in separate chapters 
later on in this manual.

Enumerated types
Often a programmer faces the need to declare a variable that will take on one of the 
small set of values. For example, a program to process astronomical observations 
might require a variable whose values represent the nine planets of the solar system. 
One possibility is to declare the variable to be of type cardinal and to use a code 
for each planet:

constant Mercury = 0u,
         Venus   = 1u,
         Earth  = 2u,
         Mars   = 3u,
         Jupiter = 4u,
         Saturn  = 5u,
         Uranus = 6u,
         Nepruve = 7u,

         Pluto   = 8u;
declare planet1, planet2 : cardinal;
constant ClosestToEarth = Venus;

SPL provides a better alternative: to declare an enumerated type with nine values, 
representing the nine planets:

type Planets = (Mercury, Venus, Earth, Mars, Jupiter,
                Saturn, Uranus, Neptune, Pluto);
declare planet1, planet2 : Planets;
constant ClosestToEarth = Venus;

Creating an enumerated type instead of using the code has several advantages. First, 
an enumerated type is easier to declare than the values of a code. Second, defining an 
enumerated type avoids the risk that two code names might accidentally be given the 
same value. Most importantly, when the programmer uses code values (of type 
cardinal, say), the compiler cannot check that values are assigned only to 
variables of the right type. For example, suppose that values of type cardinal
represent the planets, as sketched earlier. After the declaration:

constant Sun = 0u,
         . . . 
         Sat = 6u;
declare day : cardinal;// Value should be one of Sun..Sat

the programmer might forget the name of the constant representing Saturn and write:

planet1 = Sat;

This statement is perfectly legal from SPL point of view: only when the program 
produces the wrong results will the programmer become aware of any problem. 
Trying to track down an error of this nature can be difficult; using an enumerated type 
allows the compiler to detect an error.

To define an enumerated type, the programmer simply invents names for the values of 
the type, then encloses the list of these names in parentheses. The names in the list 
must all be different; furthermore, no name may be the same as the name of any other 
entity declared at the outer scope (the reason for this restriction will be explained 
below). Specifically, if two different enumerated types are declared, no value 
belonging to both can have the same name. The following declarations are illegal, 
because the identifiers Red and Blue appear in two enumerated types:

type FlagColors    = (Red, White, Blue),
     PrimaryColors = (Blue, Red, Yellow);

Enumerated types have numerous uses. Specifically, they are useful for defining 
“state” variables – variables that record the current state of a program. For example, 
the following variable allows a program to keep track of whether it is adding entries 
to a table or retrieving them from a table:

declare mode : (Enter, Fetch);

Enumerated types are also useful for expressing a result computed by a function:

type CompareResult = (Less, Equal, Greater);
procedure compare(a, b : integer) : CompareResult;

or supplying information to a procedure:

type EchoMode = (On, Off);
procedure readChar(ch : out character, mode : EchoMode);

The latter use deserves special attention. Many procedures have parameters that can 
take on only two values; the programmer may feel tempted to declare such a 
parameter to have type boolean. However, using a two-value enumerated type 
provides better program documentation. The call

readChar(ch, Off);

is much more understandable than the call

readChar(ch, false);

which would occur if the second parameter were of type boolean.

Enumerated types and scopes
If the same enumerated type (i.e. the same list of names enclosed in parentheses) is 
used in different scopes, all these uses refer to the same enumerated type. In the 
following example:

declare color : (Red, Green, Blue);

procedure proc1()
{
    declare c : (Red, Green, Blue);
   c = color;
}

the assignment in the body of proc1 is entirely legal, because the global variable 
color and the local variable c are of the same type.

To achieve that effect, SPL insists that, whenever a programmer uses an enumerated 
type, all names representing possible values of that enumerated types are implicitly 
introduced as constants at the global scope. Naturally, the type of these constants is 
exactly the enumerated type in question.

Naming convention for enumerated types
As mentioned before, all enumerated types must have values with different names. 
However, sometimes the “most natural“ names for values of different enumerated 
types are the same, as in:

type Radix = (Bin, Oct, Dec, Hex);
type Month = (Jan, Feb, Mar, Apr, May, Jun, 
              Jul, Aug, Sep, Oct, Nov, Dec);

One of the obvious ways to resolve the above situation (which is an error, because the 
value Dec is used by two different enumerated types) is to give each value full name 
instead of the abbreviated name. However, SPL coding guidelines recommend the 
other resolution – prefixing the name of each enumerated value with a name of the 
enumerated type this value belongs to. As a consequence, the two above type 
declarations will become:

type Radix = (Radix::Bin, Radix::Oct, 
              Radix::Dec, Radix::Hex);
type Month = (Month::Jan, Month::Feb, Month::Mar,
              Month::Apr, Month::May, Month::Jun, 
              Month::Jul, Month::Aug, Month::Sep,
              Month::Oct, Month::Nov, Month::Dec);

Of course, prefixing does not prevent the programmer from using full names for 
enumerated values (such as Radix::Decimal and Month::December in the 
above case), but helps to resolve the problem that would arise if the full (i.e. 
unabbreviated) names of the enumerated values are the same.

Conversion of enumerated values
In the specification of an enumerated type, the order in which the identifiers 
(enumeration values) appear is important. Each value has an ordinal number, 
representing its position in the list. The first value in the list has ordinal number 0, the 
second has ordinal number 1, and so forth. The maximum number of values in an 
enumerated type is 232-1, which reflects the fact that a variable of any enumerated 
type always occupies exactly 4 bytes of memory (technically, all enumerated values 
are represented at run time as corresponding cardinal*4 ordinal numbers).

An attempt to convert an enumerated value to any other type behaves in exactly the 
same way as if the cardinal value of the corresponding ordinal number was being 
converted:

Expression Value

integer(Mercury) 0
real(Mars) 3.0

character(Pluto) '\x08'

The inverse conversion – from a value of some other type to an enumerated value –
results in an enumerated value with the specified ordinal number:

Expression Value

Planets(0) Mercury
Planets(3.0) Mars

Planets('\x08') Pluto
Planets(100) -

The last of the above examples requires an explanation. Normally, the conversion 
Planets(100) would result in an enumerated value of the Planets type which 
has the ordinal number 100 – only there is no such value, as the type only has 9 
values. As a result, the conversion returns a value that does not belong to the type 
Planets, even though it looks as it does. This kind of errors is especially difficult to 
track; however, the SPL compiler usually provides some means of checking 
enumerated values for correctness. These checks, which, in general case, have to be 
performed at run time, slow the program down somewhat, while providing the 
guarantee that all values of enumerated types are legal. Typically, these checks are 
enabled during program development but disabled for the final (release) build; 
however, these checks can also be left enabled in the release build if the code is not 
performance – critical.

Incrementing and decrementing enumerated values
The standard operators ++ and -- can be applied to variables of enumerated types:

type Month = (Month::Jan, Month::Feb, Month::Mar,
              Month::Apr, Month::May, Month::Jun, 
              Month::Jul, Month::Aug, Month::Sep,
              Month::Oct, Month::Nov, Month::Dec);
declare month : Month = Month::Jan;
month++;

The result of such an application is, as one may suggest, the value with the previous 
(--) or next (++) ordinal number. Naturally, this can cause the value to become 
invalid (for example, if the increment “month++;“ in the above example were to be 
replaced with a decrement “month--;”, the outcome will be the variable month no 
longer containing a valid enumerated value). The run time enumerated value checks, 
as described above, will catch all situations like these.

Comparing enumerated values
Values of an enumerated type can be compared against other values of the same 
enumerated type. The comparison is based on the ordinal numbers of enumerated 
values.

As a result, enumerated values can be used anywhere a comparison is required:

for (declare p : Planets = Mercury; p <= Pluto; p++)
{
   . . . do something with p

}

Writing values of enumerated types
Because the programmer creates enumerated types, the programmer is also 
responsible for providing procedures to read and write enumerated values. 
Fortunately, few programs need to perform input and output of enumerated types in a 
human-readable form; in any event, I/O procedures for enumerated types are easy, if 
tedious, to write.

To write enumerated values in a human-readable form, a program may use a switch
statement to select a string representing a particular value. For example, the following 
procedure writes values of type Planets:

procedure writePlanet(planet : Planets)
{
    switch (planet)
    {
        when Mercury:
            Stdout::print("Mercury"); break;
            break;
        when Venus:
            Stdout::print("Venus"); break;
            break;
        when Earth:
            Stdout::print("Earth"); break;
            break;
        when Mars:
            Stdout::print("Mars"); break;
            break;
        when Jupiter:
            Stdout::print("Jupiter"); break;
            break;
        when Saturn:
            Stdout::print("Saturn"); break;
            break;
        when Uranus:
            Stdout::print("Uranus"); break;
            break;
        when Neptune:
            Stdout::print("Neptune"); break;
            break;
        when Pluto:
            Stdout::print("Pluto"); break;
            break;
    }
}

Alternatively, one can use the fact that ordinal numbers are always assigned to 
enumerated values consecutively starting with 0 (this solution uses arrays and 
pointers, both of which are discussed in more detail later on):

constant Range = 100;
type Measurements = array [2 * Range] of real;

To select an array element, the programmer writes an expression in brackets after the 
name of an array. For example, if a is the array declared earlier, we can access the 
fifth component of a by writing any of the following: a[4] (array indices start with 
0, remember), a[i] (if i has the value 4) or a[2 + 2 * j] (if j has the value 1). 
Each of these is a designator; the expression inside the brackets is the index 
expression. A designator of an array element has all of the properties of a variable; in 
particular, it can appear in the left side of an assignment operator:

for (declare i : integer = 0; i < 10; i++)
{
    a[i] = 0; // Assign 0 to each element of a
}

In an array designator the type of the index expression must be one of the integer
or cardinal types (in the above example, the index expression has the type 
integer). Furthermore, the value of the index expression must lie within the bounds 
of the array. Unless the index expression is a constant, an SPL compiler cannot check 
this condition; instead, the SPL compiler can be instructed to perform the test as the 
program executes.

SPL allows references to whole arrays (as opposed to array elements) in several 
contexts. First, an array name (representing the whole array) may appear on either 
side of an assignment operator:

a = b;

Such an assignment copies each element of b to the corresponding element of a. 
Second, the programmer may supply an array name as an argument in a procedure 
call:

sort(a);

In either case, the usual rules of type compatibility apply. For a programmer to assign 
one array to another, the arrays must have the same element types and dimensions. An 
array used as a call argument must be assignment-compatible with the corresponding 
procedure parameter.

As SPL uses structural equivalence for all types (unless specified otherwise on a per-
type basis), using a type specification instead of a type name does not affect the use of 
the array. For example, the declarations:

type ArrayType = array [10] of integer;
declare a, b : ArrayType;

can be rewritten as:

argument, without regard for its actual dimension. In most cases, the actual 
number of array elements must be passed in as a separate parameter in order 
for the procedure to know how many elements the array actually has. Note that 
in parameters and function return values cannot be free arrays, because in 
both cases the entire array must be copied, but a dimension of the free array is 
not known.

SPL disallows array elements to be free arrays; therefore, in a multidimensional array 
only the first dimension can be omitted:

type ListOfMatrices = array [][50][100] of integer;

When a free array is assigned to a bound array (or vice versa) the dimension of the 
bound array is used to determine how many elements to copy.

A free array may result from a conditional operator in which the clauses yield arrays 
with the same element type but different dimensions. Given the declarations:

declare x : integer,
        a : array [10] of integer,
        b : array [20] of integer;

the expression

(x > 0) ? a : b

yields a value of type array [] of integer, since, in general case, it is not 
known whether the result will consist of 10 or 20 elements.

As the number of elements in a free array is not known, it is impossible to perform 
full range check for array indices even at run time (it’s still possible to perform partial 
range check, however, such as ensuring that integer indices are not negative).

Conversion of array values
Just like for any other type, it is possible to convert values of other types to values of 
array types. The difference is in that the array type usually has more than one element, 
so the corresponding number of values is required for conversion.

For example, consider the following code:

type ArrayType = array [10] of integer;
declare a : ArrayType;
a = ArrayType(0, 0, 0, 0, 0, 0, 0, 0, 0, 0);

The type cast takes 10 values of type integer (which happen to be all zeroes) and 
returns a value of type array [10] of integer, which contains 10 elements 
with zero values. As is the general rule of SPL, arbitrary expressions can be used 
instead of simple constants to supply arguments for an array cast:

type ArrayType = array [3] of integer;

                                        of character;
                     middleInitial       : character;
                 };
    };

The names of structure fields are local to a particular structure type; they do not 
conflict with any other names in the program. In particular, we may use the same 
identifier to name fields in more than one structure type. Thus, the same program may 
declare both Item and Personnelrecord types, even though they both contain a 
field named number. The same field name cannot appear twice in the same structure, 
however, except within a nested structure. For example, it would be legal to add a 
field named number to NameRecord, but not to PersonnelRecord.

One advantage of structures is the increased flexibility they offer for handling a 
collection of related data values. Much of the time, a program will access individual 
fields in a structure; at other times, it is more convenient to work with the structure as 
a whole. SPL allows both forms of access.

A reference to a structure field consists of a structure name and a field name, 
separated by a period. For example, suppose that we declare the variables item1 and 
item2 as:

declare item1, item2 : Item;

To reference the number field in item1, we would write:

item1.number

One way to change the value of a structure variable is to assign values to its fields:

item1.number = 143u;
item1.description = "Men's blue umbrella";
item1.price = 9.99;
item1.onHand = 10u;

In general, a structure field of type T has the same properties as a variable of type T; it 
can appear in expressions, on the left side of an assignment, as a for loop control 
variable, and so on.

Whole structures may appear in assignment statements and as parameters in 
procedure calls; SPL allows no other whole-structure operations. Instead of assigning 
variables piecemeal to the fields of a structure, a program may instead change all 
fields by the whole-structure assignment:

item2 = item1;

This assignment copies item1.number to item2.number, 
item2.description to item1.description, and so on. Assigning one 
structure to another requires that the structures be assignment-compatible. Similarly, 

when we supply a structure as a parameter in a procedure call, the structure’s type 
must be assignment-compatible with the type of the procedure parameter. To be 
assignment-compatible, two structure types must have the same fields of the same 
type declared in the same order and they must either both have packed modifier or 
they must both lack one.

Union types
A union type is a special case of a structure type which is used when, of all the 
structure fields, one and only one is ever used at any given time. This allows the SPL 
compiler to overlap these fields, thus resulting in a substantial decrease in the 
structure size.

For example, suppose we are developing a graphical editor that allows used to draw 
individual points, lines, circles and rectangles in the integer coordinate space. 
Naturally, it’s easy enough to define a structure type for holding the properties of each 
shape:

type
    Point = structure
        {
            x, y        : integer;
        },
    Line = structure
        {
            start       : Point;
            end         : Point;
        },
    Circle = structure
        {
            center      : Point;
            radius      : cardinal;
        },
    Rectangle = structure
        {
            topLeft     : Point;
            bottomRight : Point;
        };

Since the user can create arbitrary shapes in an arbitrary order, we will, most 
probably, need a data structure that describes any of the above shapes. Here is one 
way to define such a data structure:

type
    ShapeType =
        ( 
            ShapeType::Point,
            ShapeType::Line,
            ShapeType::Circle,
            ShapeType::Rectangle
        ),

    Shape = structure
        {
            shapeType           : ShapeType;
            pointProperties     : Point;
            linePropertirs      : Line;
            circleProperties    : Circle;
            rectangleProperties : Rectangle;
        };

The Shape structure contains a field shapeType that tells what type of shape is 
actually being represented, plus the ability to store properties of any particular shape.

The problem with the above solution – although it will work just well – is serious 
over-use of memory – each Shape structure will only ever use one of the 
…Property fields, yet the memory is always reserved for all four of these.

It is in such situation – when we know that only one field of many will ever be used –
that unions are useful. The above declaration will be rewritten as:

type
    ShapeType =
        ( 
            ShapeType::Point,
            ShapeType::Line,
            ShapeType::Circle,
            ShapeType::Rectangle
        ),
    ShapeProperties = union
        {
            pointProperties     : Point;
            linePropertirs      : Line;
            circleProperties    : Circle;
            rectangleProperties : Rectangle;
        },
    Shape = structure
        {
            shapeType           : ShapeType;
            properties          : ShapeProperties;
        };

When rewritten in this manner, the type ShapeProperties can store one and only 
one of the shape properties; as a consequence, the four fields of this union can overlap 
each other in memory, thus significantly reducing the size of every Shape structure.

A thorough reader might notice that the use of the union in the above example leads to 
longer code. In a situation where, without a union being used, one would write:

declare s : Shape;
.  .  .
if (s.shapeType == ShapeType::Point)

    {
        x : integer;
        y : constant integer;
        z : volatile integer;
    };

In this situation, every reference to a field (such as s.y or s.z, provided s is of type 
StrangeStructure) behaves exactly as a normal variable with correspondingly 
modified type.

The constructs like the one above, where individual fields are assigned modified 
types, are not frequently used. They are more a result of SPL orthogonality than 
anything resulting from a practical use.

Casting to structure types
Just like for any other type, it is possible to convert values of other types to values of 
structure types. The difference is in that, like array types, structure types usually have 
more than one element, so the corresponding number of values is required for 
conversion. Naturally, each cast argument must conform to the type of the 
corresponding structure field.

For example, consider the following code:

type Point = structure { x, y : integer; };
declare p : Point;
p = Point(0, 0);

The type cast creates a new value of type Point, with both x and y fields having 
zero values, and assigns that value to the variable p. If the structure has fields of 
different types, cast arguments must be adjusted correspondingly:

type Color = (Color::Red, Color::Green, Color::Blue);
type ColoredPoint = structure 
{
    x, y : integer; 
    color : Color;
};
declare p : ColoredPoint;
p = ColoredPoint(1, 2, Color::Red);

All structure fields, whether named or anonymous, must have corresponding cast 
arguments. The only exception is a free array field (which can be specified as a last 
field of a structure), which never has a corresponding cast argument.

Casting to union types
Casting to union types is subtly different from casting to structure types, even though 
both types define a number of fields. Since a union can only store one of its fields at 
any given time, a cast to a union type must always have exactly one parameter, which 
must be assignment-compatible with one of the union’s fields. The result of such a 
cast is a union in which the field of the corresponding type is currently stored.

Consider the following code:

type Value = union (i : integer; r : real; }
declare v1, v2 : Value;
v1 = Value(1);
v2 = Value(2.3);

The cast Value(1) results in a union value whose integer field i is defined, 
while the cast Value(2.3) results in a union value whose real field r is defined 
instead. Therefore, after the above code, v1.i == 1 and v2.r == 2.3, but 
v1.r and v2.i are both undefined.

Example
Although grouping related variables into a single structure is of some value,  the most 
important advantage of structure types is their usefulness in creating other types. For 
example, we can use a structure type, such as:

type String = array [20] of character,
     Item = structure
     {
         number      : cardinal;
         description : String;
         price       : real;
         onHand      : cardinal;
     };

as the component type for an array; the resulting data structure can store information 
about many items of merchandise. We might also create a pointer type (described 
below) whose values are pointers to structures of type ITEM; a program could use this 
type to dynamically create as many ITEM records as needed.

For example, if we assume that a store stocks no more than 1000 different items, we 
could represent the entire inventory by an array of 1000 structures of type ITEM:

constant MaxInventory = 1000u;
declare inventory : array [MaxInventory] of Item,
        numItems  : cardinal; //  0..999

One operation that we might perform on such an array is to change the number of 
items on hand to reflect sales and incoming shipments. The following procedure 
performs this operation:

procedure updateInventory(
        itemNum : cardinal,
        change  : integer,
        found   : out boolean
    )
{

    for (declare currentPos : cardinal = 0u,
         currentPos < numItems;
         currentPos++)
    {
        if (itemNum == inventory[currentPos].number)
        {
            if (change >= 0)
            {
                inventory[currentPos].onHand +=
                    cardinal(number);
            }
            else
            {
                inventory[currentPos].onHand -=
                    cardinal(-number);
            }
            found = true;
            return;
        }
    }
    found = false;
}

The updateInventory procedure uses expressions such as 
inventory[currentPos].number and 
inventory[currentPos].onHand in which an index selects an array element 
and then a field identifier selects a field within an element (because the array elements 
are structures).

SPL allows arbitrary nesting of composite types.

Pointer types
Some programs are difficult to write using data structures of fixed sizes. For example, 
consider a program that maintains a list of names and telephone numbers, allowing 
the user to add new names at any time and to delete old names. Because the program 
does not know the maximum number of names in advance, it cannot store the names 
and numbers in an array. In such a program, dynamic storage allocation (and 
deallocation) is useful. Dynamic storage allocation is the allocation of storage during 
the execution of a program, in response to requests by the program. The program just 
described would perform both dynamic allocation and deallocation. When a user 
wants to add a new name to the list of names and numbers, the program requires 
enough space to store the name and associated phone number. When the user removes 
a name from the list, the program can release the space used for that name and phone 
number and can then use it for a new name (or for some other purpose). The space 
occupied by names and phone numbers will vary during execution of the program, 
increasing when the user adds a name to the list and decreasing when the user deletes 
one.

In SPL, variables can be created dynamically as a program executes. Because these 
variables are not declared in the program, they have no names; they are anonymous. A 

language that provides anonymous variables must provide some way to refer to these 
variables so that we can assign values to them and later retrieve the values. SPL 
provides pointers for this purpose. A pointer is a reference to an anonymous variable.

Defining pointer types
A pointer variable is not free to point to any anonymous variable; it must point only to 
anonymous variables of a single type (despite not having a name,  each anonymous 
variable has a type). We must indicate this type in the specification of the pointer 
type.

For example, suppose that a program is to dynamically allocate anonymous instances 
of type Complex:

type Complex = structure
    {
        realPart, imaginaryPart : real;
    };

The following declaration creates a type whose values are pointers to structures of 
type Complex:

type PtrToComplex = ^Complex;

We can then use PtrToComplex to declare pointer variables:

declare ptr1, ptr2 : PtrToComplex;

The variables ptr1 and ptr2 do not take on complex values; instead the value of 
each is a pointer to an anonymous variable of type Complex. SPL allow ptr1 and 
ptr2 to point to ordinary program variables as well as to anonymous (dynamically 
allocated) variables. Furthermore, ptr1 and ptr2 are bound to a single type 
(Complex in our example) – they cannot point at anonymous (or program) variables 
of other types (we can circumvent these restrictions when we use the predefined type 
pointer for low-level programming).

We do not have to define a named pointer type (PtrToComplex in our example), 
but can instead write:

declare ptr1, ptr2 : ^Complex;

The null pointer
A special null pointer value is guaranteed not to point to any program object. This 
value is specified by the keyword null and can be used anywhere a pointer value is 
expected.

While the null itself is of predefined type pointer (described later on in this 
manual), SPL permits automatic casting of null to any programmer-defined pointer 
type. This allows null to be assigned to pointer variables and compared to pointer 
values. For example:

declare ptr1, ptr2 : ^Complex;
ptr1 = ptr2 = null;

declares two pointer variables and assigns null to both of them (the same effect could 
have been achieved with variable initial value specifications described later on).

Creating anonymous variables
We can draw the pointer variable ptr1 as a box containing a pointer to an 
anonymous variable. Before we have assigned ptr1 a value, it is undefined:

In SPL, pointer variables (like variables in general) are not guaranteed any particular 
initial value unless one is specified by the programmer. Programs that fail to initialize 
pointer variables may produce the wrong results or fail during execution.

To create an anonymous variable, a program must use the predefined operator new
and specify the type of the anonymous variable to create. The operator new returns a 
pointer to the newly created anonymous variable or null if the anonymous variable 
cannot be creates for some reason (usually this means that there is not enough 
memory).

ptr1 = new Complex;

After the above statement the ptr1 points to a newly created anonymous Complex
structure. 

Note that the anonymous variable is not initialized during its creation.

A variation of the new operator allows creating an array of anonymous variables:

ptr1 = new Complex[3];

The above statement causes 3 anonymous Complex structures to be created, one just 
after another, and the pointer to the first of them to be assigned to ptr1:

The anonymous variables created in this way are located one after another in memory, 
without any padding in between.

Obtaining address of a named variable
For any variable v of type T, its address can always be obtained by an expression &v; 
the result is always a pointer of type ^T. This also applies to anonymous variables as 
well as named variables, so the following program fragment:

declare p1 : ^integer,
        p2 : ^integer;
p1 = new integer;
p2 = &(p1^);

causes p1 and p2 to point to the same anonymous variable. Of course, in this specific 
case the same effect could have been achieved as:

declare p1 : ^integer,
        p2 : ^integer;
p2 = p1 = new integer;

(as you may remember, in SPL assignment is a binary operator that yields a value, 
specifically – the value being assigned).

Pointer operations
After an anonymous variable is created, the programmer can assign value to it in the 
following way:

ptr1^.realPart = 5.0;
ptr1^.imaginaryPart = 1.2;

The expression ptr1^ serves as a name for the structure that ptr1 points to; the 
programmer can reference both fields within this structure by using the field names as 
selectors. After the assignment to the fields of ptr1^, it has the following 
appearance:

Dereferencing

The process of forming a name for an anonymous variable by appending the symbol ^
to a pointer value is called dereferencing. Dereferencing is one of the many operations 
that we can use to form a designator. If p is of type ^T, we can use the designator p^
in exactly the same way as any other variable of type T, specifically including its use 
as a control variable in a for loop, etc.

A special case of dereferencing can be applied when a pointer variable points to a 
structure or union. The dereferencing operator ^ and field-selection operator . (dot) 
can be combined into a single indirect field access operator ->:

ptr1->realPart = 5.0;

Another potential danger of the delete statement originates from the fact that SPL 
allows pointers to be null, to point to named program variables, or even to contain a 
value that does not point to any variable, whether named or anonymous (e.g. 
^integer(1) yields a pointer value of type ^integer, yet it is unlikely that there 
exists some variable at address 1). In all these cases, an outcome of an attempt to 
delete such pointers is undefined, although “good” SPL implementations can try to 
provide meaningful diagnostics or, at least, limit the damage.

Example: using pointers to construct linked lists
Pointers and dynamic storage allocation support the construction of linked structures. 
A linked structure is a collection of nodes, each of which contains one or more 
pointers to other nodes. A node is just a structure in which one or more fields are 
pointers.

The simplest linked structure is a linked list. A linked list consists of a number of 
nodes, each of which contains one or more pieces of data plus a pointer; the pointers 
cause the nodes to form a chain:

A pointer variable points to the first node of the list, the first node points to the second 
node, and so forth.

To illustrate linked lists, we show how to define and manipulate a linked list whose 
nodes contain single characters. First, we must declare two types: CHAR_NODE, the 
type of a node of the list, and CHAR_LIST, the type of the pointer to one of these 
nodes. A value of type CHAR_NODE will be a structure with two fields: data (the 
character stored in the node) and pNext (a pointer to the next node in the list):

type CharList = ^CharNode,
     CharNode = structure
     {
         data  : character;
         pNext : CharList;  // Points to the next node
     };

The declaration of CharList contains a forward reference to CharNode. This is 
OK, as SPL generally allows all declarations, in this example – type declarations, to 
appear in any order.

We can draw a value of type CharNode as a box with two cells named data and 
pNext:

Using this notation, a list of characters would look like this:

The variable pFirst (of type CharList) points to the first node on the list (of type 
CharNode). The list is empty (has no nodes) if pFirst is null.

We can modify the structure of a linked list in two ways: (1) by inserting a new node 
into the list and (2) by deleting a node from the list. We now look at each of these 
operations in turn.

The task of inserting a node can be separated into two cases, depending on whether or 
not we intend to insert the node after an existing node or at the beginning of the list 
(so that is becomes the first node).

Suppose first that we need to insert a node containing the character ch after an 
existing node that the variable p points to. We first create a new node  by allocating 
space for an anonymous variable  of type CharNode:

declare pNewNode : CharNode;
pNewNode = new CharNode;

We then store the desired character in the node:

pNewNode->data = ch;

We next set the pointer in the new node to point to the node following p^:

pNewNode->pNext = p->pNext;

Finally, we set the pointer in p^ to point to the new node:

p->pNext = pNewNode;

The figure below shows the result when we insert a node containing 'e' into a list 
that contains 'a', 'b' and 'd'. We insert the new node immediately after the node 
containing 'b':

The procedure insertAfter performs the operations just described:

procedure insertAfter(
        p  : CharList,

        ch : character
    )
//  Creates a node containing ch and adds it to the list
//  just after the node pointed to by p.
//  p must be non-null.
{
    declare pNewNode : CharNode;

    pNewNode = new CharNode;
    pNewNode->data = ch;
    pNewNode->pNext = p->pNext;
    p->pNext = pNewNode;
}

The operation of inserting a node at the beginning of a list is only slightly different. 
After creating a new node and storing the desired data in it, we make the new node 
point to the first node of the list, then make pFirst point to the new node, as the 
following procedure shows:

procedure insertAtFront(
        pFirst : in out CharList,
        ch     : character
    )
//  Creates a node containing ch and adds it to the front
//  of the list pointed to by pFirst.
//  pFirst may be null.
{
    declare pNewNode : CharNode;

    pNewNode = new CharNode;
    pNewNode->data = ch;
    pNewNode->pNext = pFirst;
    pFirst = pNewNode;
}

We can also divide into two cases the operation of deleting a node from a list, 
depending on whether or not another node precedes the node we want to delete. In the 
first case, suppose that p points to the node preceding the one we want to delete. We 
first set a temporary pointer (stored in the variable pOld) to point to the node to be 
deleted. We then alter the pointer in p^ to point to the node following pOld^. 
Finally, we dispose of the deleted node. For example, suppose we delete the second 
node from a list containing the characters 'a', 'b' and 'c'. The next figure shows 
the appearance of the list immediately before we destroy pOld^:

The procedure deleteAfter performs the operation just described:

procedure deleteAfter(
        p : CharList
    )
//  Deletes the node immediately following p.
//  p must be non-null.
{
    declare pOld : CharList;

    pOld = p->pNext;
    p->pNext = pOld->pNext;
    delete pOld;
}

The operation of deleting the first node on the list is similar:

procedure deleteAtFront(
        pFirst : in out CharList
    )
//  Deletes the first node.
//  pFirst must be non-null.
{
    declare pOld : CharList;

    pOld = pFirst;
    pFirst = pOld->pNext;
    delete pOld;
}

Before deleting a node, both deleteAfter and deleteAtFront are careful to 
save a pointer to the next node in the list. Consider what would happen if 
deleteAtFront tried to delete the first node without saving a pointer to the second 
node:

delete pFirst;
pFirst = pFirst->pNext;

Because delete may have changed a value of pFirst^ (filling it with garbage, for 
example, to emphasize the fact that the memory is unused – this is something many 
language implementations can do when a program is being debugged), the second 
statement may not have the desired effect.

The predefined type ‘pointer’
In addition to pointer types described so far, SPL provides one more pointer type – a 
typeless pointer type designated by the keyword pointer.

Unlike other pointer types, that always point to values of specific types (i.e. 
^integer always points to integer values), no assumptions are made as to the 
type of value pointed to by pointer (pointer in SPL is roughly equivalent to 









119

However, if one does not need parameter names in procedure types, they can be 
replaced with asterisks:

type ProcType = procedure(* : real, * : in out integer);
type FuncType = procedure(* : cardinal) : cardinal;
type AnotherProcType = procedure(*, *, * : real);

(The last of the above declarations declares a type AnotherProcType, that can 
store any procedure with three in parameters of type real).

SPL guaranteed that a value of the procedure type occupies exactly the same amount 
of memory as a pointer. Therefore, conversions between procedure types, pointer 
types (whether typed or typeless), integer and cardinal types are all lossless 
both ways, as all these conversions are performed by a simple re-interpretation of the 
value bits.

Procedure assignment
We can give a value to a procedure variable by assignment. The right side of an 
assignment may be a procedure name, a procedure variable, a call to a function that 
returns a procedure value, i.e. any expression that yields a procedure value. In this 
respect, a procedure name acts rather like a constant of a procedure type, whose value 
is a procedure with the said name.

For example, suppose that func1 and func2 are variables of type BINARY_OP:

type BinaryOp = procedure(*,* : cardinal) : cardinal;
declare func1, func2 : BinaryOp;

If the function intExp has the heading

procedure intExp(base, exponent : cardinal) : cardinal

we can assign it to func1:

finc1 = intExp;

(The absence of parentheses following the name intExp signals the compiler that 
intExp itself – not the value returned by a call of intExp – is to be assigned to 
func1). We can also copy the procedure reference stored in func1 to func2:

func2 = func1;

When a name of the procedure appears on the right side of an assignment to a 
procedure variable, the heading of the procedure must match the specification of the 
variable’s type in every detail (except, as already mentioned, for procedure parameter 
names, which are by themselves not a part of the procedure type signature). When a 
procedure variable is assigned to another procedure variable, the types of both 
variables must be the same.

Unlike other programming languages, SPL allows procedures that are local to other 
procedures to be used as procedure values, i.e. assigned to procedure variables, used 
as parameters or returned by functions. However, these cases are treated in a manner 
subtly different from top-level procedures, as described later on in this manual.

Calling a procedure variable
We can call a procedure variable once it has a value (generally, we can call any 
expression that yields a procedure value; calling a procedure directly by name is a 
specific case of calling a procedure expression, only in this case the procedure 
expression is a constant expression, i.e. a procedure name). Thus, after the assignment 
to func1 and func2, the following statements are equivalent:

Stdout::print("{cardinal}\n", intExp(i, j));
Stdout::print("{cardinal}\n", func1(i, j));
Stdout::print("{cardinal}\n", func2(i, j));

The first statement calls intExp directly. The second calls intExp indirectly 
through the variable func1, which contains a reference to intExp. And, because 
the value of func1 was copied to func2, the call of func2 in the third statement is 
also a call of intExp. A procedure variable has no particular initial value; attempting 
to call a procedure variable before we have assigned it a value will have unpredictable 
results.

Procedures used as parameters
Procedure types make it possible to supply a procedure as a parameter to a call of 
another procedure (or, maybe, even to a call of itself). For example, the following 
function, when we give it a function f as a parameter, finds the smallest non-negative 
integer n for which f(n) is 0. The parameter f must be a function that has a single 
in parameter of type integer and produces an integer result:

procedure findZero(
        f : procedure(* : integer) : integer
    ) : integer
//  Returns the smallest non-negative integer n for
//  which f(n) == 0.
//  Does not terminate if f(n) is never 0.
{
    for (declare n : integer = 0; ; n++)
    {
        if (f(n) == 0)
        {
            return n;
        }
    }
}

If poly is a following function::

procedure poly(n : integer) : integer
{
    return n * n + n -12;
}

then the call

Stdout::print("{integer}\n", findZero(poly));

writes the number 3, because poly(0) == -12, poly(1) == -10, poly(2) 
== -6 and poly(3) == 0.

Functions that return procedure values
SPL functions may return values of a procedure type. The following function, given 
one of the characters '+', '-', '*' or '/', returns a function to perform the 
corresponding arithmetic operation:

type ArithmeticOp = procedure(*, * : real) : real;
procedure FindOp(operator : character) : ArithmeticOp
{
    switch (operator)
    {
        when '+':
            return add;
        when '-':
            return subtract;
        when '*':
            return multiply;
        when '/':
            return divide;
    }
}

add, subtract, multiply and divide are programmer-defined functions of 
type ArithmeticOp.

Example: Displaying a file in Decimal, Hex or Octal
Some programs need to choose a procedure from a group of several alternatives. One 
way to do so is to use a variable of an unstructured type to keep track of which 
procedure to call. For example, suppose that we are writing a program that will 
display the ordinal numbers of the characters in a file in either base 8, base 10 or base 
16. The program prompts the user to enter a letter d, h or o to indicate a decimal, 
hexadecimal or octal format. Each time it writes a character, the program uses an if
or switch statement to determine the output format:

switch (base)
{
    when 'd':
        printDecimal(buffer[i]);

        break;
    when 'h':
        printHexadecimal(buffer[i]);
        break;
    when 'o':
        printOctal(buffer[i]);
        break;
}

where the printing procedures can be defined as:

procedure printDecimal(c : character)
{
    Stdout::print(" {cardinal|%1d}", cardinal*1(c));
}

procedure printHexadecimal(c : character)
{
    Stdout::print(" {cardinal|%1x}", cardinal*1(c));
}

procedure printOctal(c : character)
{
    Stdout::print(" {cardinal|%1o}", cardinal*1(c));
}

After the initial assignment to base its value does not change during the execution of 
the program; consequently, the program wastes some time by repeatedly testing the 
value of base against 'd', 'h' and 'o'. By using a procedure variable, the 
program can avoid making the test.

We first observe that that PrintDecimal, PrintHexadecimal and 
PrintOctal are all compatible with the following type:

type PrintProcedure = procedure(* : character);

Therefore, we can declare a variable capable of storing either printDecimal, 
printHexadecimal or printOctal:

declare printNumber : PrintProcedure;

Once the user has selected decimal, hex or octal, a switch statement stores either 
printDecimal, printHexadecimal or printOctal in printNumber. 
When the time comes to write the ordinal number of a character, the statement:

printNumber(c);

will call printDecimal, printHexadecimal or printOctal.

The following program does what we have just described. After asking the user for a 
base (decimal, hex or octal) and a file name, the program reads the file character-by-
character and writes the ordinal numbers of these characters to the screen. Note that 
the program uses console I/O routines for user interaction and file I/O routines for 
reading the file; all these routines are part of the SPL standard library.

#include "conio.sph"
#include "fcntl.sph"
#include "io.sph"
using spl;

procedure main() public
{
    type PrintProcedure = procedure(* : character);

    declare printNumber : PrintProcedure,
            fileName    : array[MaxPathLength] of
                          character;

    procedure printDecimal(c : character)
    {
        Console::print("{cardinal*1|%8u}", cardinal*1(c));
    }

    procedure printHexadecimal(c : character)
    {
        Console::print("{cardinal*1|%8.2X",
                       cardinal*1(c));
    }

    procedure printOctal(c : character)
    {
        Console::print("{cardinal*1|%8.3o",
                         cardinal*1(c));
    }

    procedure askForBase()
    {
        declare base : character;
        Console::print("Display in Dec, Hex or Octal? ");
GetBase:
        for (; ; )
        {
            if (not Console::readCharacter(base))
            {
                continue;
            }
            switch (base)
            {
                when 'd':
                when 'D':

                    printNumber = printDecimal;
                    break GetBase;
                when 'h':
                when 'H':
                    printNumber = printHexadecimal;
                    break GetBase;
                when 'o':
                when 'O':
                    printNumber = printOctal;
                    break GetBase;
            }
        }
        Console::print("{character}\n", base);
    }
    
    procedure askForFileName()
    {
        for (; ; )
        {
            Console::print("Enter name of input file: ");
            if (Console::scan("{string}", fileName) == 1)
            {
                return;
            }
        }
    }

    procedure printFile()
    {
        declare f : File,
                c : character;

        f = File::open(fileName,
                       File::OpenMode::OpenExisting,
                       File::AccessMode::Read,
                       File::ShareMode::Read);
        if (f == File::None)
        {
            Console::print("Cannot read the file\n");
            return;
        }
        while (File::readCharacter(f, c))
        {
            printNumber(c);
        }
        closeFile(f);
    }

    askForBase();
    askForFileName();
    printFile();



125

}

Storing a procedure in a variable works well in the above case because the variable 
does not change after the initial assignment. In a program that calls one of several 
procedures inside a loop, using a procedure variable saves no time, because the 
program would still need an if or switch statement. In such situations – when a
program must choose among a group of procedures based on the value of a variable n
– we may find it worthwhile to store the procedures in an array and use n as an index 
to select an appropriate procedure.

declare xx : array [] of integer = 
             array [3] of integer (1, 2, 3);

In such situation, the actual number of elements in the variable’s initial value is used 
to convert variable from a free array to bound array; therefore, in the above example, 
the actual type of xx will be array [3] of integer, not array [] of 
integer.

Since explicit casts of array element lists to free array types actually yield bound 
arrays as their resulting value, the above declaration can be rewritten as:

declare xx : array [] of integer = 
             array [] of integer (1, 2, 3);

Inferring variable types from initial values
Since an initial value of a variable always has a specific type, it is not necessary to 
specify the type of a variable which has an initial value. In this case, the variable is 
assumed to be of exactly the same type as its initial value, i.e. the declaration:

declare x : public = 3;

is exactly the same as:

declare x : public integer = 3;

An SPL compiler, however, is permitted to issue a warning when a variable is 
declared without an explicit type. This reflects the fact that omitting the type from a 
variable declaration is not generally recommended, as it makes the program more 
prone to type errors.

Specifying variable linkage
For statically allocated variables, SPL allows the programmer to specify the exact 
details of how the linker sees the variable. This is not a feature generally required for 
application programming; however, in low-level system programming the ability to 
define the low-level details of a variable allocation comes in handy.

Aliasing variables
Any statically allocated variable, regardless of its visibility, can be assigned an alias.

A variable alias is an arbitrary string that is used in place of a variable name in all 
linker error messages that need to mention the said variable. For example, consider 
the following code fragment:

declare iobuf : external array [1024] of cardinal*1 
                alias "I/O buffer";

When a compilation unit containing the above declaration is linked with other 
compilation units to form an executable module, it may well happen that the actual 
definition for the iobuf variable does not exist in any of these compilation units. In 

    {
        writeChar(digits[i]);
    }
}

Specific variants can then be used to direct output to a different location:

procedure printNumberToFile(
        n : cardinal,
        f : File
    )
{
    procedure writeChar(c : character)
    {
        File::writeCharacter(f, c);
    }
    printNumber(n, writeChar);
}

procedure printNumberToStdout(
        n : cardinal
    )
{
    procedure writeChar(c : character)
    {
        Stdout::writeCharacter(c);
    }
    printNumber(n, writeChar);
}

procedure PrintNumberToString(
        n : cardinal,
        s : ^character
    )
{
    procedure writeChar(c : character)
    {
        (s++)^ = c;
    }
    printNumber(n, writeChar);
    s^ = '\x00';
}

Note the use of non-local reference to parameter f within writeChar in 
printNumberToFile. The same trick is used in printNumberToString, only 
here the current output position is incremented within writeChar each time a 
character is written; the final value of the output position pointer is then used to write 
the NUL character terminating a string.

type SomeType = packed structure
    {
        c : character;
        n : integer;
        d : cardinal*2;
    };

will lay out values of SomeType in the following manner, regardless of the target 
platform:

Anonymous fields
Normally, when a programmer defined a structure or union type, he must specify 
names and types for all structure or union fields. However, SPL allows structure or 
union fields to be introduced without a name. Such fields are called anonymous fields.

Although anonymous fields have no name and, therefore, cannot be accessed, they 
turn out to be useful in several situations.

Building variant data structures
If an anonymous structure or union field is itself a structure or union, then the fields 
of the inner structure or union can be accessed as if they were members of the original 
structure. For example, consider the declaration:

type
    Point = structure
        {
            x, y        : integer;
        },
    Line = structure
        {
            start       : Point;
            end         : Point;
        },
    Circle = structure
        {
            center      : Point;
            radius      : cardinal;
        },
    Rectangle = structure
        {
            topLeft     : Point;
            bottomRight : Point;
        },
    ShapeType =
        ( 
            ShapeType::Point,

            ShapeType::Line,
            ShapeType::Circle,
            ShapeType::Rectangle
        ),
    ShapeProperties = union
        {
            pointProperties     : Point;
            linePropertirs      : Line;
            circleProperties    : Circle;
            rectangleProperties : Rectangle;
        },
    Shape = structure
        {
            shapeType           : ShapeType;
            properties          : ShapeProperties;
        };

declare shape : Shape;

which defines a data structure capable of storing various shapes in terms of their 
coordinates. Although useful, the above declaration forces the programmer to use 
long chains of qualifications to access shape properties:

shape.properties.pointProperties.x = 0;

Although these long qualification chains do not cause the code to be inefficient, they 
are tedious to write. To simplify things, one can notice that the properties field is 
never accessed on its own because it’s the sub-fields we are interested in. Declaring 
the field anonymous:

type
    Shape = structure
        {
            shapeType           : ShapeType;
            *                   : ShapeProperties;
        };

does not alter the layout of the data structure, but allows shorted qualification chains 
to be written:

shape.pointProperties.x = 0;

The SPL compiler notices that, although a Shape structure does not itself have a 
pointProperties field, it nevertheless contains an anonymous union which does 
have a pointProperties field, so that union’s field us used.

Note that all qualifications must be unambiguous. Therefore, in the following code 
fragment:

type

    Shape = structure
        {
            shapeType           : ShapeType;
            *                   : ShapeProperties;
            *                   : ShapeProperties;
        };

shape.pointProperties.x = 0;

the assignment is an error because SPL compiler sees two anonymous union fields, 
each containing a sub-field named pointProperties, and is unable to choose 
between them.

To shorten things even more, one can notice that all fields in structures Point, 
Line, Circle and Rectangle have different names, and introduce even more 
anonymous fields:

type
    ShapePropertires = union
        {
            *                   : Point;
            *                   : Line;
            *                   : Circle;
            *                   : Rectangle;
        },
    Shape = structure
        {
            shapeType           : ShapeType;
            *                   : ShapeProperties;
        };

In this case, one can access the shape properties with even shorter qualification 
chains:

switch (shape.shapeType)
{
    when ShapeType::Point,
        shape.x = shape.y = 0;
        break;
    when ShapeType::Line,
        shape.start = shape.end = Point(0, 0);
        break;
    when ShapeType::Circle,
        shape.center = Point(0, 0);
        shape.radius = 0;
        break;
    when ShapeType::Rectangle:
        shape.topLeft = shape.bottomRight = Point(0, 0);
        break;
}

Indirect qualification in the presence of anonymous fields is performed in exactly the 
same manner:

declare pShape : ^Shape;
   .   .   .
pShape->shapeType = ShapeType::Point;
pShape->x = pShape->y = 0;

Specifying structure layout
When a format of the data the program must process is defined outside the scope of 
the program, the programmer usually must use a packed structure to match the 
required format exactly. However, it sometimes happens that the required data format 
contains within it bytes that are unused, or of no interest to the program.

In these cases, a structure field must be specified to span these unused bytes. 
Declaring this field without a name eliminates the risk that the program will 
accidentally try to access the unused data.

For example, consider the page tables on Intel 80286 processor. Each page table entry 
is 8 bytes long, but two of these bytes are unused:

type PageTableEntry = packed structure
    {
        segmentSize             : cardinal*2;
        segmentAddressAndAccess : cardinal*4;
        *                       : cardinal*2;
    };

Note that the structure is packed – without the modifier the SPL compiler could 
have introduced padding bytes between the structure fields to make access more 
efficient.

Label types
As its name suggests, a value of a label type represents a label associated with some 
statement. The label type is denoted by the SPL keyword label:

declare lb : label;

Label operations
Any label can be assigned to a label variable:

declare lb : label;

lb = X;
X : Stdout::print("Hello!\n");

In this respect, statement labels act as constants of a label type.

In addition, a label variable can be assigned a null value:

declare lb : label;

lb = null;

Although null itself is of type pointer, SPL allows automatic casting of null to 
label type. The result is a label value that is guaranteed not to refer to any 
statement in the SPL program,

Label values can be compared for equality or inequality (but not for order such as 
“less-then”):

declare lb : label;

lb = X;
X : Stdout::print("Hello!\n");
Y : Stdout::print("Goodbye!\n");
if (lb == X) .   .   .
   .   .   .
if (lb <> Y) .   .   .
   .   .   .
if (lb <> null) .   .   .

The last statement once again illustrates the fact that SPL performs implicit casting of 
nulls to label type whenever necessary.

Indirect jumps

Any expression yielding a label result can be used as a target of the goto
statement. Therefore, in the following fragment:

declare lb : label;

lb = X;
goto lb;
   .   .   .
X : Stdout::print("Hello!\n");

the statement “goto lb;” has the same effect as “goto X;” would have had in its 
place.

Like any other values, label values can be used to form more complex data structures. 
For example, consider the following fragment:

declare b      : boolean,
        labels : array[2] of label;

labels[0] = X;
labels[1] = Y;
   .   .   .
goto labels[cardinal(b)];

    ) : EvalResult
{
    .   .   .
    if (<a syntax error is detected>)
    {
        return EvalResult::SyntaxError;
    }
    .   .   .
    if (<division by zero detected>)
    {
        return EvalResult::DivisionByZero;
    }
    .   .   .
    value = <calculated value>;
    return EvalResult::Success;
}

While the above approach is perfectly suitable, the caller of the Eval procedure must 
check for the returned error code in every place Eval is called. What’s even worse, 
the caller is free to completely ignore the returned error code, whether intentionally or 
by mistake, thus disabling proper error handling.

An alternative strategy for error handling is for the caller to supply the label(s) to 
jump to in case of an error:

procedure Eval(
        pExpressionString     : ^constant character,
        syntaxErrorHandler    : label,
        divisionByZeroHandler : label
    ) : real
{
    .   .   .
    if (<a syntax error is detected>)
    {
        goto syntaxErrorHandler;
    }
    .   .   .
    if (<division by zero detected>)
    {
        goto divisionByZeroHandler;
    }
    .   .   .
    return <calculated value>;
}

In this case, the caller has no choice but to provide error handling facilities, which can 
be shared for multiple calls to eval:

   .   .   .
x = eval(expression1, 

         HandleSyntaxError,
         HandleDivisionByZero);
   .   .   .
y = eval(expression1, 
         HandleSyntaxError,
         HandleDivisionByZero);
   .   .   .
HandleSyntaxError:     // The syntax error handler
   .   .   .
HandleDivisionByZero:  // The division by zero handler
   .   .   .

Derived types
As has been already mentioned, SPL uses structural equivalence for all types by 
default. As a consequence, in the following code fragment:

type String = array [80] of character,
     Name = structure
            {
                firstName : String;
                lastName  : String;
            };
declare n1 : Name,
        n2 : structure
            {
                firstName : String;
                lastName  : String;
            },
        n3 : structure
            {
                firstName : array [80] of character;
                lastName  : array [80] of character;
            };

the variables n1, n2 and n3 all have the same type and are all assignment-
compatible. Generally, the type declaration in SPL does not introduce new types, but 
merely creates named aliases to existing types.

Sometimes, however, the programmer requires exactly the opposite – to make a type 
stand apart in order to avoid accidentally mixing it with other types. SPL allows this 
by declaring derived types.

A declaration of a derived type looks just like a declaration of a “normal” named type 
with one difference – a keyword new preceding the type declaration means that a 
derived type is being declared:

type Weight = new real,
     String = new array [80] of character,
     Name  = new structure
              {

p = pi;  // OK: ^integer -> pointer is an implicit
         // type conversion
p = ai;  // OK: array [10] of integer -> pointer is
         // an implicit type conversion
p = npi; // Error: derived type Npi cannot be
         // implicitly converted to anything, even
         // though it is semantically a pointer.
p = nai; // Error: derived type Nai cannot be
         // implicitly converted to anything, even
         // though it is semantically an array.

The fact that the derived types are different from other types provides a type-safe way 
to implement many typical constructs.

Example: Handles
In a modular programming, the situation frequently occurs when a module or a library 
provides means of access to objects it implements by means of abstract “handles”. A 
typical example is a file I/O, which allows the program to open a file, returning an 
abstract “file handle”, which must then be used for all I/O operations involving the 
file in question.

Many system programming languages (including C) just say “handles are integer 
values” and leave it at that. Although this approach results in fairly efficient code, the 
programmer is not protected from accidentally storing the handle in an integer 
variable which is then used in arithmetic operation. Moreover, if in a call to a file I/O 
procedure an arbitrary integer value is accidentally specified instead of a proper 
handle, the problem goes unnoticed. Other typical errors include handle type 
mismatch, such as specifying a file handle where a window handle should have been 
used instead.

SPL rectifies all these situations by allowing each handle type to be declared as a 
separate derived type. Here is how a “file handle” type and file I/O API can be 
defined:

type File = new integer;
constant File::None = File(-1);

procedure openFile(
        fileName   : ^constant character,
        .   .   .
    ) : File;

procedure writeFile(
        file         : File,
        .   .   .
    ) : boolean;

In the presence of the above declarations, the file I/O can be performed in a clear and 
concise manner:

declare f : File;
f = openFile(fileName, .   .   .);
if (f == File::None)
{
   .   .   . // Handle the file open error
}
writeFile(f, .   .   .);

However, any attempt to use an integer value, or a handle of something other than a 
file, for file I/O will be spotted by the compiler:

type Window = new integer; // Another handle
declare w : Window;
   .   .   .
writeFile(1, .   .   .); // Error: integer is not a File
writeFile(w, .   .   .); // Error: Window is not a File

Opaque types
When creating large programs, breaking them into more or less independent modules 
is a standard way to make programming easier and improve program maintainability. 
One of the most important features of such modular design is encapsulation – if a 
module implements some data types and/or services, the client code shall be able to 
use these types/services without having to know details of their implementation and, 
in an optimal case; these details shall be totally hidden from the client.

To give an example of how encapsulation works, suppose we a writing an application 
for tracking items available in a warehouse. Naturally, we need a data structure to 
describe these items:

type Item = structure
     {
         number      : cardinal;
         description : array [20] of character;
         price       : real;
         onHand      : cardinal;
     };

To complement that, we need a data structure that stores the current contents of the 
warehouse:

constant MaxItems = 1024;
declare warehouseItems     : array [MaxItems] of Item,
        warehouseItemCount : cardinal;

One way to write the warehouse management program is to directly access the above 
two variables each time a warehouse contents must be examined or changed. This, 
however, quickly leads to a situation where the same sequences of code for accessing 
warehouse data are repeated all over the program, as typical warehouse operations 
(such as “check if an item is in stock”, etc.) will be needed in many places.

To improve the situation, one may consider writing dedicated procedures for 
implementing frequently needed warehouse services. The code below shows how two 
of such procedures could have been implemented: the one that checks if a desired 
number of a specific item is in stock, another to find a warehouse item by its code and 
return it:

procedure isItemInStock(
        number           : cardinal,
        requiredQuantity : cardinal
    ) : boolean
{
    declare pItem : ^Item = findItem(number);
    if (pItem <> null)
    {
        return pItem->onHand >= requiredQuantity;
    }
    return false;
}

procedure findItem(
        number      : cardinal,
    ) : ^Item
{
    for (declare n : cardinal = 0u; 
         n < warehouseItemCount;
         n++)
    {
        if (warehouseItems[n].number == number)
        {
            return &warehouseItems[n];
        }
    }
    return null;
}

Although using dedicated access procedures for warehouse data simplifies the 
programming, there is still a danger of the program accidentally or deliberately 
accessing the warehouse data directly, bypassing the access procedures.

To hide the implementation details, SPL provides a concept of opaque types. 

An opaque type is a type for which only a type name is known, but nothing else. To 
declare an opaque type, the standard type declaration directive is used, with the type 
signature left out:

type Item;

Since nothing is known about an opaque type except for its name, a program cannot 
declare variables (or parameters, or structure/union fields, etc.) of opaque types:

declare item : Item; // Error

What is permitted is declaring pointers to opaque types:

declare pItem : ^Item; // OK

When writing a modular SPL program that uses opaque types, the following steps are 
taken:

1. A dedicated SPL header file is created to define the public interface of a 
module. For the above example, this header file can be named 
warehouse.sph and written among the following lines:

type Item;

procedure isItemInStock(
        number           : cardinal,
        requiredQuantity : cardinal
    ) : boolean;

procedure findItem(
        number      : cardinal,
    ) : ^Item;

Any program that wishes to use the public warehouse interface must include 
the warehouse.sph header using the SPL preprocessor.

2. The SPL file providing the actual implementation of the warehouse module 
can be named warehouse.spl and will refine the definition of the Item
type as well as providing implementations of the public warehouse procedures
(the public modifier on the procedures makes them visible in other 
compilation units, i.e. includes them into the public interface of the warehouse 
module):

#include "warehouse.sph"

type Item = structure
     {
         number      : cardinal;
         description : array [20] of character;
         price       : real;
         onHand      : cardinal;
     };
constant MaxItems = 1024;
declare warehouseItems : array [MaxItems] of Item,
        warehouseItemCount : cardinal;

procedure isItemInStock(
        number           : cardinal,
        requiredQuantity : cardinal
    ) : boolean public
{

    declare pItem : ^Item = findItem(number);
    if (pItem <> null)
    {
        return pItem->onHand >= requiredQuantity;
    }
    return false;
}

procedure findItem(
        number      : cardinal,
    ) : ^Item public
{
    for (declare n : cardinal = 0u; 
         n < warehouseItemCount;
         n++)
    {
        if (warehouseItems[n].number == number)
        {
            return &warehouseItems[n];
        }
    }
    return null;
}

Note that within the warehouse.spl implementation file the type Item is 
not opaque – even though the header file warehouse.sph gives an opaque 
definition of the type Item, this definition is then locally refined, defining the 
Item to actually be a structure. Note also that the variables 
warehouseItems and warehouseItemCount are private to the 
implementation file, and, therefore, are safe from being accessed from other 
compilation units.

3. Any program that wants to use the warehouse module must include the 
warehouse.sph header. Once this is done, the program can use the 
procedures declared in the warehouse.sph header to access the warehouse:

if (findItem(itemNumber) == null)
{   // There are no items with the specified number
   //  in the warehouse
   .   .   . 
}

Note that, since type Item is opaque as far as the client program is concerned, 
the warehouse data cannot be accessed directly. In particular, even when a 
warehouse item is found, the program cannot access its price directly, but must 
use the dedicated procedure from the warehouse interface:

if (pItem->price > availableAmount)        // Error
    .   .   .

if (getItemPrice(pItem) > availableAmount) // OK

            bottomRight : Point;
        },
    ShapeType =
        ( 
            ShapeType::Point,
            ShapeType::Line,
            ShapeType::Circle,
            ShapeType::Rectangle
        ),
    ShapeProperties = union
        {
            pointProperties     : Point;
            linePropertirs      : Line;
            circleProperties    : Circle;
            rectangleProperties : Rectangle;
        },
    Shape = structure
        {
            shapeType           : ShapeType;
            *                   : ShapeProperties;
        };

declare shape : Shape;
shape.shapeType = ShapeType::Circle;
shape.circleProperties.center = Point(0, 0);
shape.circleProperties.radius = 1;

In the above sequence of assignments, one can notice that the same structure shape
is qualified over and over again. In order to avoid having to write all these 
qualifications, a with statement can be used:

declare shape : Shape;
with (shape)
{
    .shapeType = ShapeType::Circle;
    .circleProperties.center = Point(0, 0);
    .circleProperties.radius = 1;
}

The header of the with statement specifies a structure or union that is being repeatedly 
qualified. Within the body of the with statement, all implicit qualifications, such as 
.shapeType, are resolved by using the structure or union mentioned in the with 
statement header.

Implicit qualification works just as well with pointers to structures or unions:

declare pShape : Shape;
with (pShape)
{
    ->shapeType = ShapeType::Circle;

    ->circleProperties.center = Point(0, 0);
    ->circleProperties.radius = 1;
}

The above statement can be shortened even further, if one notices that the value -
>circleProperties is repeatedly qualified in the body of the with statement:

declare pShape : ^Shape;
with (pShape)
{
    ->shapeType = ShapeType::Circle;
    with (->circleProperties)
    {
        .center = Point(0, 0);
        .radius = 1;
    }
}

If several with statements are nested into each other and some implicit qualification 
in the body of an inner with statement can be resolved in more than one way, the 
resolution provided by the innermost relevant with statement is always used.

Apart from making programs shorter, with statements come very handy when a 
structure or union value being qualified in an r-value (i.e. not a variable). Consider the 
following example:

procedure getShape() : Shape;

.    .    .

if (getShape().shapeType == ShapeType::Circle)
{
    Stdout::print("Circle: center = 
({integer},{integer}), radius = {cardinal}\n",
           getShape().circleProperties.center.x,
           getShape().circleProperties.center.y,
           getShape().circleProperties.radius);
}

The problem with the above code is, of course, in that the getShape() is called 
multiple times. The first solution that comes to mind is using a Shape variable to 
capture the getShape() return value:

declare shape : Shape = getShape();
if (shape.shapeType == ShapeType::Circle)
{
    Stdout::print("Circle: center = 
({integer},{integer}), radius = {cardinal}\n",
           shape.circleProperties.center.x,
           shape.circleProperties.center.y,

    {
        goto divisionByZeroHandler;
    }
    .   .   .
    return <calculated value>;
}

The primary advantage of this approach is in that any procedure can define its own, 
local, error handling facilities, without the need to modify the code of the procedure 
where the non-local jump is actually performed. Consider the following procedure:

procedure eval(
        pExpressionString : ^constant character
    ) : real
{
    //  Definitions used for local error handling
    declare savedSyntaxErrorHandler    : label,
            savedDivisionByZeroHandler : label;

    procedure setupErrorHandlers()
    {
        savedSyntaxErrorHandler = syntaxErrorHandler;
        savedDivisionByZeroHandler = divisionByZeroHandler;
        syntaxErrorHandler = HandleSyntaxError;
        divisionByZeroHandler = HandleDivisionByZeroError;
    }

    procedure restoreErrorHandlers()
    {
        syntaxErrorHandler = savedSyntaxErrorHandler;
        divisionByZeroHandler = savedDivisionByZeroHandler;
    }

    // Set up local error handlers
    setupErrorHandlers();

    //  Call eval(). Note that the non-local jumps
    //  from eval() will activate local error handlers
    //  defined within Eval1().
    declare value : real = eval(pExpressionString);

    //  Restore error handlers & return
    restoreErrorHandlers();
    return value;

    //  Error handlers:
HandleSyntaxError:
    Stdout::print("Syntax error”);
    RestoreErrorHandlers();
    return 0.0;

1. Procedure preconditions are evaluated in the same order they are written. If 
some of the preconditions is false, the contract is failed and an error is 
reported.

2. All values used as arguments of old pseudo-function in procedure 
postconditions are evaluated and saved for future use.

3. The procedure body is executed.
4. Procedure postconditions are evaluated in the same order they are written. If 

some of the postconditions is false, the contract is failed and an error is 
reported.

Naming contract elements
Any pre- and post-condition can be assigned a name, which is written before the 
require/ensure keyword and separated from it by a colon:

procedure sqrt(x : real) : real
    SqrtArgumentIsNonNegative : require
        x >= 0.0;
    SqrtResultIsCorrect : ensure
       absR((result * result) – x) < Epsilon;
{
    .    .    . // Evaluate & return a square root of x
}

Depending on the SPL compiler, the name of the pre- or post-condition may be 
included into the error message issued when the pre- or post-condition is violated. If 
these names are chosen to describe the conditions they are associated with, this may 
help in finding out what exactly went wrong.

Interface and implementation contracts
SPL allows pre- and post-conditions to be specified in both procedure declarations 
and procedure implementations. Consider the following SPL code:

//  Header file myfunc.sph
procedure myfunc(p : pointer) : pointer
    ArgumentNotNull: require
        p <> null;
    ResultNotNull: ensure
        result <> null;

//  Implementation file myfunc.spl
procedure myfunc(p : pointer) : pointer
    GoesForward: ensure
        result >= p;
{
    //  Implementation of the desired behaviour
}

Note that the contract of the myfunc procedure is split between the procedure 
declaration that appears in the header file and the procedure definition that appears in 

the language or the compiler more complicated. This section describes SPL support 
for namespace emulation.

Defining common name prefixes
It is an SPL convention that names of all program elements that belong together (for 
example, Names that constitute a public API of a reusable component) shall be made 
to use the same prefix, separated from the program element name by double-colon. 
For example, if we are developing a reusable library for handling variable-length 
character strings, we may choose the prefix “strlib” for all public API elements of 
that library:

type strlib::String;              // Opaque "String" type
procedure strlib::initialize();   // Initializes strlib
procedure strlib::deinitialize(); // Performs cleanup
procedure strlib::getStringLength(
        str : ^strlib::String
    ) : cardinal;                 // Queries the length 
                                  // of the string
//  And so on...

These prefixes are not limited to one; an arbitrary number of prefixes can be applied 
to a name in order to model nested namespaces. Note, however, that an entire name, 
including all its prefixes and double-colon separators, is lexically a single identifier 
and cannot, for example, include spaces.

Using prefixed names
A name that includes one or more prefixes can be used just as any ordinary identifier. 
Returning to the above example, a string library client may well contain the code like 
the following:

procedure printStringLength(
        str : ^strlib::String
    )
{
    Stdout::print("length = {cardinal}",
                 strlib::getStringLength(str));
}

All in all, the client code that uses prefixed names looks very much like a code in a 
much higher-level language (such as C++) that always uses fully qualified names.

Default prefixes and name lookup rules
Often, it is too cumbersome to always write full forms of prefixed names, especially 
when these names have several prefixes or name prefixes are long. To avoid the need 
to always write names in their full form, SPL allows specifying default prefixes, 
which act as name lookup rules.

A default prefix specification has the form:

using <identifier> [ , <identifier> ...] ;

declare s1::x : integer;
declare s1::s2::y : integer;
declare s2::z : integer;
using s1;
using s2;
.  .  .
x = 0;         // OK, s1::x is in scope
y = 0;         // ERROR, s1::s2::y is not cumulative
s2::y = 0;     // OK, s1::s2::y is in scope
s1::s2::y = 0; // OK, full qualification
z = 0;         // OK, s2::z is in scope

As for all other SPL declarations, using directives do not have to be effected before 
use, so the following SPL code is legal:

declare s1::x : integer;
declare s1::s2::y : integer;
.  .  .
x = 0;         // OK, s1::x is in scope
s2::y = 0;     // OK, s1::s2::y is in scope
s1::s2::y = 0; // OK, full qualification
z = 0;         // OK, s2::z is in scope
using s2;
using s1;

although, of course, this is not the recommended coding style.

Specifying default libraries
When building reusable components, it is often desirable that any code that uses the 
certain reusable component is automatically linked against the implementation of the 
said component. This practice is called default library pulling and works by 
specifying, in the header file that defined the component’s API, the name of the 
library where the component implementation resides.

Other languages, such as C and C++, leave default library pulling to the mercy of a 
particular implementation (typically, with a compiler-specific pragmas such as 
“#pragma comment("lib", <library name>)” in Microsoft Visual C++).
However, the ability to pull in libraries was found to be so important to component-
based development, that SPL provides default library pulling as a standard language 
features.

The default library pulling in SPL is implemented via the use directive:

use <library name> [ , <library name> …];

where each <library name> specifies the name of one library that shall be pulled 
in by the linker. There is no need to pull in the standard SPL runtime library, as this 
one is pulled in by default (unless a dedicated SPL compiler command-line option is 
used NOT to pull in the standard SPL library – this can happen when writing OS 

The SPL preprocessor
SPL implements the C-style preprocessor. Each preprocessor directive occupies a 
single line and has the hash ‘#’ in the leftmost non-space position. A line with 
backslash as last character is assumed to continue to the next line; any number of 
continuation lines is allowed.

The preprocessor directive is a directive keyword (which always starts with ‘#’ and is 
case-sensitive) followed by whatever parameters are required by directive. Both C-
style ( /* … */ ) and C++ - style ( // ) comments are allowed in preprocessor 
directives and are replaced with a single space per comment. The C-style comment in 
a preprocessor directive must end in the same line where it begins.

SPL preprocessor allows any number of spaces between the initial ‘#’ and the 
directive keyword. However, it is a convention to write the two without spaces in 
between.

#include

#include <filename>

The <filename> must be a valid SPL string constant (ASCII, Unicode and ISO 
10646 strings are allowed, and ASCII strings can be specified without the prefix ‘A’). 
However, some implementations may prohibit Unicode or ISO 10646 (or both) 
filenames. ASCII filenames are always valid. Standard SPL syntax for string 
constants is used (i.e. if a host OS uses backslash ‘\’ as a path separator, the 
<filename> will look like “sys\\types.h” – specifying the filename 
“sys\types.h” will cause the ‘\t’ to be interpreted as escape sequence 
representing ASCII TAB character). This is different from what C preprocessor does.

If <filename> specified full path, this full path is used to locate the include file.

If <filename> specifies partial path, the SPL compiler looks for include file in the 
following order:

1. Current directory.
2. Directory where the source file containing the ‘#include’ directive is 

located, then up the inclusion stack to the root.
3. Standard include paths, if specified.

The body of the ‘#include’ directive is subject to macro replacement.

This directive also has a qualified form:

#include <filename> from <include context name>

An include context name is syntactically an identifier. Each include context represents 
a list of directories. Compiler searches for the include file ONLY in the directories 

designated by the specified include context and nowhere else (specifically, standard 
include paths are ignored, as are paths containing the file where the qualified 
#include itself appears). How include contexts are defined is platform-specific (for 
example, on Win32 platforms environment variables are used for that purpose). The 
standard include context SPL_STANDARD_LIBRARY_HEADERS is used for SPL 
standard library headers.

#once

#once

When encountered, the directive causes the preprocessor to remember the full file 
name of the file in which the ‘#once’ has occurred. Later an attempt to ‘#include’ 
the same file will locate the file as usual but then ignore it instead of including it. The 
directive is devised to prevent multiple inclusions of the same header.

The position of the #once directive in the file is irrelevant; the file is marked as a 
“once file, and already included at that” as soon as #once is encountered. A file can 
contain any number of #once directives.

#define

#define <name> <replacement string>
#define <name>(<parameters>) <replacement string>

This directive defined a specified macro. The parameters list, if present, can be empty,
in which case a macro must be called with an empty parameter list.

If a macro has parameters, these parameters can appear in a macro body, the 
corresponding macro call will cause the parameters to be replaced with actual call 
arguments. If a macro parameters is preceded with ‘#’ or ‘#@’ in the macro body, it 
is, correspondingly, stringized (i.e. converted into an ASCII string constant) or 
charized (i.e. converted into an ASCII character constant). The characters ‘##’ in the 
macro body are removed, thus allowing for token-pasting.

If a macro is already defined, its previous definition is removed and the new one takes 
effect.

SPL also allows defining macros with variable number of parameters, which have the 
three-dots ‘…’ as the last (or even only) element of their parameter list:

#define p(f,...) Stdout::print((f), __VA_ARGS__)

Within the body of such variadic macro, the special placeholder __VA_ARGS__ is 
replaced with a list of all unmatched macro arguments during macro expansion. For 
example, the statement:

p("{integer} {integer} {integer}\n", x, y, x + y);

is macroexpanded by SPL preprocessor to:

Stdout::print("{integer} {integer} {integer}\n",
              x, y, x + y);

#undef

#undef <name>

This directive undefines a specified macro. If a macro is already not defined, the 
directive has no effect.

#error

#error <any text>

This directives causes the error message to be generated with the specified text. The 
error text is subject to macro expansion.

#warning

#warning <any text>

This directives causes the warning message to be generated with the specified text. 
The warning text is subject to macro expansion.

#message

#message <any text>

This directives causes an arbitrary message to be generated when preprocessing 
reaches the directive. The message text is subject to macro expansion. Note that, 
unlike #error or #warning directives, the #message directive does not interrupt 
compilation.

#line

#line <line number>
#line <line number> <file name>

This directive causes the compiler to replace the currently tracked source line number 
and, if specified, source file name. The directive parameters are subject to macro 
expansion.

#if, #elif, #else, and #endif

#if <preprocessor expression>
   <text>

[#elif <preprocessor expression>
   <text> …]
[#else
   <text> ]
#endif

The #if directive, with the #elif, #else, and #endif directives, controls 
compilation of portions of a source file. If the expression you write (after the #if) 
has a nonzero value, the line group immediately following the #if directive is 
retained in the translation unit.

Each #if directive in a source file must be matched by a closing #endif directive, 
which must be in the same source file as the #if directive (i.e. #if … #endif
blocks cannot cross source file boundaries). Any number of #elif directives can 
appear between the #if and #endif directives, but at most one #else directive is 
allowed. The #else directive, if present, must be the last directive before #endif.

The #if, #elif, #else, and #endif directives can nest in the <text> portions 
of other #if directives. Each nested #else, #elif, or #endif directive belongs 
to the closest preceding #if directive.

Macro replacement is performed within the part of the command line that follows an 
#if or #elif command, so a macro call can be used in the preprocessor expression.

The preprocessor selects one of the given occurrences of <text> for further 
processing. A block specified in <text> can be any sequence of text. It can occupy 
more than one line. Usually <text> is program text that has meaning to the compiler 
or the preprocessor.

The preprocessor processes the selected <text> and passes it to the compiler. If 
<text> contains preprocessor directives, the preprocessor carries out those 
directives. Only text blocks selected by the preprocessor are compiled.

The preprocessor selects a single <text> item by evaluating the constant expression 
following each #if or #elif directive until it finds a true (nonzero) preprocessor 
expression. It selects all text (including other preprocessor directives beginning with 
#) up to its associated #elif, #else, or #endif.

If all occurrences of <preprocessor expression> are false, or if no #elif
directives appear, the preprocessor selects the text block after the #else clause. If 
the #else clause is omitted and all instances of <preprocessor 
expression> in the #if block are false, no text block is selected.

The <preprocessor expression> is an integer constant expression with these 
additional restrictions: 

__FILE__
The name of the current source file. __FILE__ expands 
to a string surrounded by double quotation marks:
    "<file name>"

__FILE1__

The name of the current source file. __FILE__ expands 
to an ASCII string surrounded by double quotation 
marks:
    a"<file name>"

__FILE2__

The name of the current source file. __FILE__ expands 
to a Unicode string surrounded by double quotation 
marks:
    u"<file name>"

__FILE4__

The name of the current source file. __FILE__ expands 
to an ISO-10646 string surrounded by double quotation 
marks:
    i"<file name>"

__LINE__
The line number in the current source file. The line 
number is a decimal integer constant. It can be altered 
with a #line directive.

__DATE__ The compilation date of the current source file. The date 
is a string literal of the form "Mmm dd yyyy".

__DATE1__
The compilation date of the current source file. The date 
is an ASCII string literal of the form a"Mmm dd yyyy".

__DATE2__
The compilation date of the current source file. The date 
is a Unicode string literal of the form u"Mmm dd yyyy".

__DATE4__
The compilation date of the current source file. The date 
is an ISO-10646 string literal of the form i"Mmm dd
yyyy".

__TIME__ The compilation time of the current source file. The time 
is a string literal of the form "hh:mm:ss".

__TIME1__
The compilation time of the current source file. The time 
is an ASCII string literal of the form a"hh:mm:ss".

__TIME2__
The compilation time of the current source file. The time 
is a Unicode string literal of the form u"hh:mm:ss".

__TIME4__
The compilation time of the current source file. The time 
is an ISO-10646 string literal of the form i"hh:mm:ss".

__TIMESTAMP__

The date and time of the last modification of the current 
source file, expressed as a string literal in the form "Ddd 
Mmm Date hh:mm:ss yyyy", where Ddd is the 
abbreviated day of the week and Date is an integer from 
1 to 31.

__TIMESTAMP1__ The date and time of the last modification of the current 

source file, expressed as an ASCII string literal in the 
form a"Ddd Mmm Date hh:mm:ss yyyy", where Ddd is 
the abbreviated day of the week and Date is an integer 
from 1 to 31.

__TIMESTAMP2__

The date and time of the last modification of the current 
source file, expressed as a Unicode string literal in the 
form u"Ddd Mmm Date hh:mm:ss yyyy", where Ddd is 
the abbreviated day of the week and Date is an integer 
from 1 to 31.

__TIMESTAMP4__

The date and time of the last modification of the current 
source file, expressed as an ISO-10646 string literal in 
the form i"Ddd Mmm Date hh:mm:ss yyyy", where Ddd
is the abbreviated day of the week and Date is an integer 
from 1 to 31.

__VA_ARGS__

When used in a body of a variadic macro, this identifier
is replaced with a list of all unmatched macro arguments. 
In all other contents, __VA_ARGS__ has no special 
meaning.

Compiler-specific predefined macros
The following table summarized compiler-specific SPL predefined macros defined by 
the reference SPL compiler developed by Cybernetic Intelligence GmbH. These are 
guaranteed to be defined by the Cybernetic Intelligence GmbH SPL compiler itself, 
without a need to include any standard header.

Macro name Macro value

__COMPILER_CIGMBH__

Defined as 1 by the 
reference SPL compiler 
developed by Cybernetic 
Intelligence GmbH. Other 
compilers will define 
different symbols.

__BUILD_MODE_DEBUG__
Defined as 1 if and only if 
the compilation is performed 
in debug mode.

__BUILD_MODE_RELEASE__
Defined as 1 if and only if 
the compilation is performed 
in release mode.

__TARGET_ENVIRONMENT_HOSTED__

Defined as 1 if and only if 
the compiler generates code 
for a hosted environment; 
otherwise undefined.

__TARGET_ENVIRONMENT_FREESTANDING__ Defined as 1 if and only if 

the compiler generates code 
for a freestanding 
environment; otherwise 
undefined.

__TARGET_CPU_IA32__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code; otherwise 
undefined.

__TARGET_CPU_IA32_80386__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for 80386 or 
above; otherwise undefined.

__TARGET_CPU_IA32_80486__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for 80486 or 
above; otherwise undefined.

__TARGET_CPU_IA32_PENTIUM__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for Pentium or 
above; otherwise undefined.

__TARGET_CPU_IA32_PENTIUM2__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for Pentium 2 or 
above; otherwise undefined.

__TARGET_CPU_IA32_PENTIUM3__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for Pentium 3 or 
above; otherwise undefined.

__TARGET_CPU_IA32_PENTIUM4__

Defined as 1 if and only if 
the compiler generates the 
IA-32 code for Pentium 4 or 
above; otherwise undefined.

__TARGET_BYTE_ORDER_BIG_ENDIAN__

Defined as 1 if and only if 
the compiler generates code 
for big-endian platform; 
otherwise undefined.

__TARGET_BYTE_ORDER_LITTLE_ENDIAN__

Defined as 1 if and only if 
the compiler generates code 
for little-endian platform;
otherwise undefined.

__TARGET_OS_NONE__

Defined as 1 if and only if 
the compiler generates code 
for a bare hardware (no OS), 
otherwise undefined. This 
setting allows compiling OS 

kernels.

__TARGET_OS_WINDOWS__

Defined as 1 if and only if 
the compiler generates code 
for some Windows platform; 
otherwise undefined.

__TARGET_OS_WIN16__

Defined as 1 if and only if 
the compiler generates code 
for 16-bit Windows 
platform; otherwise 
undefined.

__TARGET_OS_WIN32__

Defined as 1 if and only if 
the compiler generates code 
for 32-bit Windows 
platform; otherwise 
undefined.

__TARGET_OS_WIN64__

Defined as 1 if and only if 
the compiler generates code 
for 64-bit Windows 
platform; otherwise 
undefined.

__TARGET_STDLIB_STATIC__

Defined as 1 if and only if 
the compiler generates code 
that will statically link to 
SPL standard library.

__TARGET_STDLIB_DYNAMIC__

Defined as 1 if and only if 
the compiler generates code 
that will dynamically link to 
SPL standard library.

__HOST_CPU_IA32__

Defined as 1 if and only if 
the compiler runs on an IA-
32 processor; otherwise 
undefined.

__HOST_CPU_IA32_80386__

Defined as 1 if and only if 
the compiler runs on 80386 
or above; otherwise 
undefined.

__HOST_CPU_IA32_80486__

Defined as 1 if and only if 
the compiler runs on 80486 
or above; otherwise 
undefined.

__HOST_CPU_IA32_PENTIUM__

Defined as 1 if and only if 
the compiler runs on
Pentium or above; otherwise 
undefined.

__HOST_CPU_IA32_PENTIUM2__

Defined as 1 if and only if 
the compiler runs on
Pentium 2 or above; 
otherwise undefined.

__HOST_CPU_IA32_PENTIUM3__

Defined as 1 if and only if 
the compiler runs on
Pentium 3 or above; 
otherwise undefined.

__HOST_CPU_IA32_PENTIUM4__

Defined as 1 if and only if 
the compiler runs on
Pentium 4 or above; 
otherwise undefined.

__HOST_BYTE_ORDER_BIG_ENDIAN__

Defined as 1 if and only if 
the compiler runs on a big-
endian platform; otherwise 
undefined.

__HOST_BYTE_ORDER_LITTLE_ENDIAN__

Defined as 1 if and only if 
the compiler runs on a little-
endian platform; otherwise 
undefined.

__HOST_OS_WINDOWS__

Defined as 1 if and only if 
the compiler runs on
Windows platform; 
otherwise undefined.

__HOST_OS_WIN16__

Defined as 1 if and only if 
the compiler runs on 16-bit 
Windows platform; 
otherwise undefined.

__HOST_OS_WIN32__

Defined as 1 if and only if 
the compiler runs on 32-bit 
Windows platform; 
otherwise undefined.

__HOST_OS_WIN64__

Defined as 1 if and only if 
the compiler runs on 64-bit 
Windows platform; 
otherwise undefined.

Conceptually, the above predefined macros are organized into a set of hierarchies, 
with one hierarchy per target platform property (these properties are COMPILER, 
TARGET_ENVIRONMENT, TARGET_CPU, TARGET_BYTE_ORDER, TARGET_OS, 
etc.)

<STRING8> ::= <A> “ [ <CHAR8> …] “

<STRING16> ::= <U> “ [ <CHAR16> …] “

<STRING32> ::= <I> “ [ <CHAR32> …] “

<STRINGX> ::= “ [ <CHAR8|CHAR16|CHAR32 depending on platform> …] “

<A> ::= a | A

<U> ::= u | U

<I> ::= i | I

<CHARACTER> ::= <CHARACTER8> | <CHARACTER16>
  | <CHARACTER32> | <CHARACTERX>

<STRING> ::= <STRING8> [ <STRING8> …]
  | <STRING16> [ <STRING16> …]
  | <STRING32> [ <STRING32> …]
  | <STRINGX> [ <STRINGX> …]

LL(k) syntax

<translation unit> ::= [ <declaration> …]

<declaration> ::= <type declaration>
  | <constant declaration>
  | <variable declaration>
  | <procedure declaration>
  | <using declaration>
  | <use declaration>

<type declaration> ::= type <typedef> [ , <typedef> …] ;

<typedef> ::= <NAME> [ = <type attribute> 
             [ <type attribute> …] ]

<type attribute> ::= <type signature>
  | deprecated
  | new

<constant declaration> ::= constant <constdef> [ , <constdef> …] ;

<constdef> ::= <NAME> = <constant attribute> 
           [ <constant attribute> …]

<constant attribute> ::= <e1>
  | deprecated

<variable declaration> ::= declare <vardef> [ , <vardef> …] ;

<vardef> ::= <name list> :
    [ <variable attribute> …]

<name list> ::= <NAME> [ , <NAME> …]

<variable attribute> ::= <type signature>
  | <memory class>
  | <visibility>

  | <alias>
  | <segment>
  | <global>
  | <initializer>
  | weak
  | deprecated

<memory class> ::= static
  | automatic
  | register
  | shared
  | thread

<visibility> ::= public
  | private
  | external
  | export
  | import [ <STRING> ]

<alias> ::= alias <STRING>

<segment> ::= segment <STRING>

<global> ::= global <STRING>

<initializer> ::= = <e1>

<procedure declaration> ::= <procedure header>
    <procedure body>

<procedure header> ::= procedure <NAME>
    <parameter list> <return type>
    [ <procedure attribute> …]

<parameter list> ::= ( [ <parameters> ] )

<parameters> ::= <parameter> [ , <parameter> …] [ , ... ]
  | ...

<parameter> ::= <optname list> :
    [ <parameter attribute> …]

<optname list> ::= <optname> [ , <optname> …]

<optname> ::= <NAME> | *

<parameter attribute> ::= <type signature>
  | <memory class>
  | <parameter direction>

<parameter direction> ::= in | out | in out

<return type> ::= [ : <type signature> ]

<procedure attribute> ::= <visibility>
  | <alias>
  | <global>
  | <segment>
  | inline
  | weak
  | deprecated

<procedure body> ::= ;
  | [ <condition> …]
    <block statement>

<condition> ::= <precondition>
  | <postcondition>

<precondition> ::= [ <NAME> : ] require <expression> ;

<postcondition> ::= [ <NAME> : ] ensure <expression> ;

<using declaration> ::= using <NAME> [ , <NAME> …] ;

<use declaration> ::= use <STRING> [ , <STRING> …] ;

<type signature> ::= <primitive type>
  | <named type>
  | <pointer type>
  | <structure type>
  | <union type>
  | <enumerated type>
  | <array type>
  | <procedure type>
  | constant <type signature>
  | volatile <type signature>
  | packed <type signature>

<primitive type> ::= integer*1
  | integer*2
  | integer*4
  | integer*8
  | integer
  | cardinal*1
  | cardinal*2
  | cardinal*4
  | cardinal*8
  | cardinal
  | real*4
  | real*8
  | real
  | character*1
  | character*2
  | character*4
  | character
  | pointer
  | boolean
  | label

<named type> ::= <NAME>

<pointer type> ::= ^ <type signature>

<structure type> ::= structure <fields>

<union type> ::= union <fields>

<enumerated type> ::= ( [ <enumeration literal list> ] )

<enumeration literal list> ::= <enumeration literal>
    [ , <enumeration literal> …]



196

<enumeration literal> ::= <NAME>

<array type> ::= array <dimensions> of <type signature>

<dimensions> ::= <dimension> [ <dimension> …]

<dimension> ::= [ [ <e1> ] ]

<fields> ::= { [ <field> …] }

<field> ::= <optname list> : <type signature> ;

<procedure type> ::= procedure <parameter list> <return type>

<ds> ::= <statement>
  | <declaration>

<statement> ::= <NAME> : <statement>
  | <empty statement>
  | <expression statement>
  | <block statement>
  | <if statement>
  | <switch statement>
  | <while statement>
  | <do statement>
  | <for statement>
  | <goto statement>
  | <break statement>
  | <continue statement>
  | <return statement>
  | <assert statement>
  | <delete statement>
  | <with statement>

<empty statement> ::= ;

<expression statement> ::= <expression> ;

<block statement> ::= {
        [ <ds> ]
    }

<if statement> ::= if ( <expression> ) <statement>
    [ else <statement> ]

<switch statement> ::= switch ( <expression> )
    {
      [ <alternative> …]
    }

<alternative> ::= <selector> [ <selector> …]
    <ds> [ <ds> …]

<selector> ::= when <expression> [ ... <expression> ] :
  | default :

<while statement> ::= while ( <expression> ) <statement>

<do statement> ::= do <statement> while ( <expression> ) ;

<for statement> ::= for ( [ <expression> ] ;
    [ <expression> ] ; [ <expression> ] )
    <statement>

<goto statement> ::= goto <expression> ;

<break statement> ::= break [ <expression> ] ;

<continue statement> ::= continue [ <expression> ] ;

<return statement> ::= return [ <expression> ] ;

<assert statement> ::= assert <expression> ;

<delete statement> ::= delete <expression> ;

<with statement> ::= with ( <expression> ) <statement>

<expression> ::= <e1> [ , <e1> …]

<e1> ::= <e2> [ <assign op> <e1> ]

<assign op> ::= =
  | <binary op> =

<binary op> ::= + | - | * | / | %
  | << | >>
  | == | <> | < | <= | > | >=
  | and [ then ]
  | implies [ then ]
  | xor
  | or [ else ]

<e2> ::= <e3> [ ? <e2> : <e2> ]

<e3> ::= <e4> [ <or op> <e4> …]

<or op> ::= or [ else ]

<e4> ::= <e5> [ xor <e5> …]

<e5> ::= <e6> [ <implies op> <e6> …]

<implies op> ::= implies [ then ]

<e6> ::= <e7> [ <and op> <e7> …]

<and op> ::= and [ then ]

<e7> ::= <e8> [ <cmp op> <e8> …]

<cmp op> ::= == | <> | < | <= | > | >=

<e8> ::= <e9> [ <shift op> <e9> …]

<shift op> ::= << | >>

<e9> ::= <e10> [ <add op> <e10> …]

<add op> ::= + | -

<e10> ::= <e11> [ <mul op> <e11> …]

<mul op> ::= * | / | %

<e11> ::= <un op> <e11>
  | <e12>

<un op> ::= + | - | not | & | ++ | --

<e12> ::= <e13> [ <e12suffix> …]

<e12suffix> ::= . <NAME>
  | -> <NAME>
  | ^
  | [ <expression> ]
  | ( [ <e1> [ , <e1> …] ] )
  | ++
  | --

<e13> ::= ( <expression> )
  | <NAME>
  | <INTEGER>
  | <REAL>
  | <CHARACTER>
  | <STRING>
  | true
  | false
  | null
  | result
  | <type signature> ( <e1> [ , <e1> …] )
  | sizeof ( <expression> )
  | sizeof ( <type signature> )
  | old ( <expression> )
  | new <type signature> [ [ <expression> ] ]
  | . <NAME>
  | -> <NAME>
  | declare <vardef>

Appendix B: GNU Free Documentation 
License

Version 1.2, November 2002 

Copyright (C) 2000,2001,2002  Free Software Foundation, Inc.
51 Franklin St, Fifth Floor, Boston, MA  02110-1301  USA
Everyone is permitted to copy and distribute verbatim copies
of this license document, but changing it is not allowed.

0. PREAMBLE

The purpose of this License is to make a manual, textbook, or other functional and 
useful document "free" in the sense of freedom: to assure everyone the effective 
freedom to copy and redistribute it, with or without modifying it, either commercially 
or noncommercially. Secondarily, this License preserves for the author and publisher 
a way to get credit for their work, while not being considered responsible for 
modifications made by others. 

This License is a kind of "copyleft", which means that derivative works of the 
document must themselves be free in the same sense. It complements the GNU 
General Public License, which is a copyleft license designed for free software. 

We have designed this License in order to use it for manuals for free software, 
because free software needs free documentation: a free program should come with 
manuals providing the same freedoms that the software does. But this License is not 
limited to software manuals; it can be used for any textual work, regardless of subject 
matter or whether it is published as a printed book. We recommend this License 
principally for works whose purpose is instruction or reference. 

1. APPLICABILITY AND DEFINITIONS

This License applies to any manual or other work, in any medium, that contains a 
notice placed by the copyright holder saying it can be distributed under the terms of 
this License. Such a notice grants a world-wide, royalty-free license, unlimited in 
duration, to use that work under the conditions stated herein. The "Document", below, 
refers to any such manual or work. Any member of the public is a licensee, and is 
addressed as "you". You accept the license if you copy, modify or distribute the work 
in a way requiring permission under copyright law. 

A "Modified Version" of the Document means any work containing the Document or 
a portion of it, either copied verbatim, or with modifications and/or translated into 
another language. 

A "Secondary Section" is a named appendix or a front-matter section of the Document 
that deals exclusively with the relationship of the publishers or authors of the 
Document to the Document's overall subject (or to related matters) and contains 
nothing that could fall directly within that overall subject. (Thus, if the Document is in 

part a textbook of mathematics, a Secondary Section may not explain any 
mathematics.) The relationship could be a matter of historical connection with the 
subject or with related matters, or of legal, commercial, philosophical, ethical or 
political position regarding them. 

The "Invariant Sections" are certain Secondary Sections whose titles are designated, 
as being those of Invariant Sections, in the notice that says that the Document is 
released under this License. If a section does not fit the above definition of Secondary 
then it is not allowed to be designated as Invariant. The Document may contain zero 
Invariant Sections. If the Document does not identify any Invariant Sections then 
there are none. 

The "Cover Texts" are certain short passages of text that are listed, as Front-Cover 
Texts or Back-Cover Texts, in the notice that says that the Document is released 
under this License. A Front-Cover Text may be at most 5 words, and a Back-Cover 
Text may be at most 25 words. 

A "Transparent" copy of the Document means a machine-readable copy, represented 
in a format whose specification is available to the general public, that is suitable for 
revising the document straightforwardly with generic text editors or (for images 
composed of pixels) generic paint programs or (for drawings) some widely available 
drawing editor, and that is suitable for input to text formatters or for automatic 
translation to a variety of formats suitable for input to text formatters. A copy made in 
an otherwise Transparent file format whose markup, or absence of markup, has been 
arranged to thwart or discourage subsequent modification by readers is not 
Transparent. An image format is not Transparent if used for any substantial amount of 
text. A copy that is not "Transparent" is called "Opaque". 

Examples of suitable formats for Transparent copies include plain ASCII without 
markup, Texinfo input format, LaTeX input format, SGML or XML using a publicly 
available DTD, and standard-conforming simple HTML, PostScript or PDF designed 
for human modification. Examples of transparent image formats include PNG, XCF 
and JPG. Opaque formats include proprietary formats that can be read and edited only 
by proprietary word processors, SGML or XML for which the DTD and/or processing 
tools are not generally available, and the machine-generated HTML, PostScript or 
PDF produced by some word processors for output purposes only. 

The "Title Page" means, for a printed book, the title page itself, plus such following 
pages as are needed to hold, legibly, the material this License requires to appear in the 
title page. For works in formats which do not have any title page as such, "Title Page" 
means the text near the most prominent appearance of the work's title, preceding the 
beginning of the body of the text. 

A section "Entitled XYZ" means a named subunit of the Document whose title either 
is precisely XYZ or contains XYZ in parentheses following text that translates XYZ 
in another language. (Here XYZ stands for a specific section name mentioned below, 
such as "Acknowledgements", "Dedications", "Endorsements", or "History".) To 
"Preserve the Title" of such a section when you modify the Document means that it 
remains a section "Entitled XYZ" according to this definition. 

The Document may include Warranty Disclaimers next to the notice which states that 
this License applies to the Document. These Warranty Disclaimers are considered to 
be included by reference in this License, but only as regards disclaiming warranties: 
any other implication that these Warranty Disclaimers may have is void and has no 
effect on the meaning of this License. 

2. VERBATIM COPYING

You may copy and distribute the Document in any medium, either commercially or 
noncommercially, provided that this License, the copyright notices, and the license 
notice saying this License applies to the Document are reproduced in all copies, and 
that you add no other conditions whatsoever to those of this License. You may not use 
technical measures to obstruct or control the reading or further copying of the copies 
you make or distribute. However, you may accept compensation in exchange for 
copies. If you distribute a large enough number of copies you must also follow the 
conditions in section 3. 

You may also lend copies, under the same conditions stated above, and you may 
publicly display copies. 

3. COPYING IN QUANTITY

If you publish printed copies (or copies in media that commonly have printed covers) 
of the Document, numbering more than 100, and the Document's license notice 
requires Cover Texts, you must enclose the copies in covers that carry, clearly and 
legibly, all these Cover Texts: Front-Cover Texts on the front cover, and Back-Cover 
Texts on the back cover. Both covers must also clearly and legibly identify you as the 
publisher of these copies. The front cover must present the full title with all words of 
the title equally prominent and visible. You may add other material on the covers in 
addition. Copying with changes limited to the covers, as long as they preserve the title 
of the Document and satisfy these conditions, can be treated as verbatim copying in 
other respects. 

If the required texts for either cover are too voluminous to fit legibly, you should put 
the first ones listed (as many as fit reasonably) on the actual cover, and continue the 
rest onto adjacent pages. 

If you publish or distribute Opaque copies of the Document numbering more than 
100, you must either include a machine-readable Transparent copy along with each 
Opaque copy, or state in or with each Opaque copy a computer-network location from 
which the general network-using public has access to download using public-standard 
network protocols a complete Transparent copy of the Document, free of added 
material. If you use the latter option, you must take reasonably prudent steps, when 
you begin distribution of Opaque copies in quantity, to ensure that this Transparent 
copy will remain thus accessible at the stated location until at least one year after the 
last time you distribute an Opaque copy (directly or through your agents or retailers) 
of that edition to the public. 

In the combination, you must combine any sections Entitled "History" in the various 
original documents, forming one section Entitled "History"; likewise combine any 
sections Entitled "Acknowledgements", and any sections Entitled "Dedications". You 
must delete all sections Entitled "Endorsements." 

6. COLLECTIONS OF DOCUMENTS

You may make a collection consisting of the Document and other documents released 
under this License, and replace the individual copies of this License in the various 
documents with a single copy that is included in the collection, provided that you 
follow the rules of this License for verbatim copying of each of the documents in all 
other respects. 

You may extract a single document from such a collection, and distribute it 
individually under this License, provided you insert a copy of this License into the 
extracted document, and follow this License in all other respects regarding verbatim 
copying of that document. 

7. AGGREGATION WITH INDEPENDENT WORKS

A compilation of the Document or its derivatives with other separate and independent 
documents or works, in or on a volume of a storage or distribution medium, is called 
an "aggregate" if the copyright resulting from the compilation is not used to limit the 
legal rights of the compilation's users beyond what the individual works permit. When 
the Document is included in an aggregate, this License does not apply to the other 
works in the aggregate which are not themselves derivative works of the Document. 

If the Cover Text requirement of section 3 is applicable to these copies of the 
Document, then if the Document is less than one half of the entire aggregate, the 
Document's Cover Texts may be placed on covers that bracket the Document within 
the aggregate, or the electronic equivalent of covers if the Document is in electronic
form. Otherwise they must appear on printed covers that bracket the whole aggregate. 

8. TRANSLATION

Translation is considered a kind of modification, so you may distribute translations of 
the Document under the terms of section 4. Replacing Invariant Sections with 
translations requires special permission from their copyright holders, but you may 
include translations of some or all Invariant Sections in addition to the original 
versions of these Invariant Sections. You may include a translation of this License, 
and all the license notices in the Document, and any Warranty Disclaimers, provided 
that you also include the original English version of this License and the original 
versions of those notices and disclaimers. In case of a disagreement between the 
translation and the original version of this License or a notice or disclaimer, the 
original version will prevail. 

If a section in the Document is Entitled "Acknowledgements", "Dedications", or 
"History", the requirement (section 4) to Preserve its Title (section 1) will typically 
require changing the actual title. 

9. TERMINATION

You may not copy, modify, sublicense, or distribute the Document except as 
expressly provided for under this License. Any other attempt to copy, modify, 
sublicense or distribute the Document is void, and will automatically terminate your 
rights under this License. However, parties who have received copies, or rights, from 
you under this License will not have their licenses terminated so long as such parties 
remain in full compliance. 

10. FUTURE REVISIONS OF THIS LICENSE

The Free Software Foundation may publish new, revised versions of the GNU Free 
Documentation License from time to time. Such new versions will be similar in spirit 
to the present version, but may differ in detail to address new problems or concerns. 
See http://www.gnu.org/copyleft/. 

Each version of the License is given a distinguishing version number. If the 
Document specifies that a particular numbered version of this License "or any later 
version" applies to it, you have the option of following the terms and conditions either 
of that specified version or of any later version that has been published (not as a draft) 
by the Free Software Foundation. If the Document does not specify a version number 
of this License, you may choose any version ever published (not as a draft) by the 
Free Software Foundation.


